
Programação em ambiente WINDOWS.

Assim como várias outras linguagens (Visual C, Visual Objects, Delphi), a linguagem VISUAL BASIC (VB) é uma linguagem de programação que visa o desenvolvimento de aplicações para o ambiente WINDOWS.

Linguagem baseada em eventos:

    Tem por objetivo gerenciar os eventos que ocorrem com os objetos na tela.

Exemplo de uma aplicação Windows:

OBJETOS: São os controles que compõem a tela, tais como botões, caixas de texto, labels, etc.

Aos objetos estão  associados: 

     PROPRIEDADES: São características inerentes aos objetos, tais como: cor, tamanho, tipo de fonte, etc. 

     EVENTOS: São as ações que se exercem sobre os objetos, tais como: click, change, keypress, etc.

     MÉTODOS: São funções e procedures internas do Visual Basic, tais como: print, move, clear, etc.

A grande vantagem do ambiente WINDOWS  é a padronização da interface com o usuário, seja qual for o programa que iremos utilizar. Dessa maneira, os programadores em ambiente WINDOWS devem conhecer os padrões dessa plataforma e de maneira alguma usar sua “criatividade” para inovações. De qualquer maneira, o próprio VB se encarrega de manter boa parte dos padrões adotados no WINDOWS:

Telas e janelas principais do Visual Basic V 5.0:

Ao darmos início à execução do ambiente de desenvolvimento do VB, surge a tela para efetuarmos a escolha do tipo de projeto que iremos desenvolver. Parte dessa tela é mostrada abaixo:

Ao escolhermos “Standard Exe”, que é onde serão concentrados os estudos deste nosso curso básico de VB, surge a seguinte tela do ambiente de desenvolvimento da linguagem Visual Basic. 

A tela principal do VB segue a tendência das interfaces conhecidas como VDI (Visual Development Interface), e é mostrada abaixo:

                                                                                         ToolBar (Barra de ferramentas)
                                                                      Barra de menu
[image: image3.png]& [=[Bx]
Ele Edt View roject Fomat Debug Run Took Addns Window Help

B-5-HE

Genera

ET—
Aphabetc | Categoried |

Farml
1-3
Fale

O asisooor
2-Siable
Form1

True

Tue
13- Copy P
0-Sold

f

Tue

Ei[lFoun Loyout ____E3]

Returnsfsets the text
displayed n an object’s ttle





                                                                                        Form Layout window
ToolBox (Caixa de Ferramentas)         
                                               Project Explorer (Janela de Projeto)
                        Form (Formulário)
                                                                                                           Properties (Janela de propriedades)
MODIFICAR A JANELA DO VB PARA QUE FIQUE COMO NA TELA ANTERIOR.

Brevíssima descrição/utilização dos componentes mostrados na página anterior:

Barra de menu: É comum a todas a aplicações Windows. Muitas de suas funções são repetidas na “ToolBar”.

ToolBar(Barra de Ferramentas): Por meio de ícones, pode-se executar rapidamente algumas funções existentes ou não na Barra de Menu. Exemplo: Start, End, Save Project, Open Project, etc.

ToolBox (Caixa de ferramentas): Possui os ícones dos “controles”, que serão incorporados ao “Formulário”, com a finalidade de compor a tela da aplicação que iremos desenvolver.

Form Layout window: Tem por objetivo permitir a visualização e posicionamento na tela do computador, dos formulários componentes do projeto

Project Explorer (Janela de Projeto): Permite a visualização e seleção dos formulários e módulos componentes do projeto.

Form (Formulário): É a tela onde são incluídos controles, que comporão a interface entre o usuário e o computador.

Properties (Janela de propriedades): Fornece a lista de propriedades do controle que está selecionado. Pode-se escolher a lista em ordem alfabética ou por categoria.

EXERCÍCIOS:

- INCLUIR UM BOTÃO NO FORMULÁRIO (FORM1);

- ALTERNAR ENTRE O FORMULARIO E O BOTÃO (diz-se que o objeto está SELECIONADO);

- OBSERVAR AS PROPRIEDADES DA “JANELA DE PROPRIEDADES”

OBS.: note que a propriedade “name” (nome do controle) é tão importante, que é a primeira da lista.

- INCLUIR UM ÍCONE (propriedade “Icon” do Formulário);

- EXECUTAR O PROGRAMA (verifique o ícone no formulário e na barra de tarefas).

SALVAR O PROJETO EM SEU DISCO DE TRABALHO:  Quando for salvar a primeira vez, é fundamental seguir o seguinte procedimento:

1 - abrir o item FILE do menu e executar o “SAVE FORM1 AS nome_do_formulário”;
2 - executar o “SAVE PROJECT AS nome_do_projeto”.
De tempos em tempos, é necessário ir salvando o projeto para preservar os incrementos efetuados no desenvolvimento no programa. Para isso, basta executar apenas o item “SAVE PROJECT” que o VB se encarrega de salvar todos os arquivos associados ao projeto.

É possível gerar o programa executável e gravá-lo em disco para uso posterior , sem mais haver necessidade do VB.

Para isso basta abrir o item FILE do menu e executar o item “MAKE nome_do_projeto.EXE”, gravando-o por exemplo em seu disquete.  Após feito isso, pode-se executar o programa diretamente pelo item “Executar” do Windows. Não há necessidade de arquivos de “runtime”, como nas versões anteriores do VB.

ETAPAS RECOMENDADAS  PARA PROGRAMAÇÃO EM VB:

1- Criação da tela (interface com o usuário);

2- Configuração das propriedades associadas aos objetos;

3- Escrever os programas em VISUAL BASIC.

Passos recomendados no item 1:

. Abrir o item “File” e escolher “New Project”;

. Salvar o Formulário e o Projeto (save form1 as, save project as);

. Incluir todos os objetos necessários na tela (controles);

. Salvar o projeto (save project).

Passos recomendados no item 2:

. Mudar a propriedade “name” de todos os objetos:

- Essa propriedade é muito importante, pois é através do “name” que os objetos são referenciados na programação em VB.

- Recomenda-se que se estabeleça um padrão para as 3 primeiras letras de cada tipo de objeto, e que o nome seja relacionado à funcão que o objeto vai desempenhar.

- Sugestão de nomes: 

* Botão de comando : CMD       ex. CMDCLICKME

* Caixa de texto : TXT   ex. TXTMENSAGEM

* Label                 : LBL   ex. LBLROTULO

* Botão de opção : OPT   ex. OPTLIGADO

. Mudar as demais propriedades para adequar os controles ao objetivo da aplicação (CAPTION, COLOR, etc.)

Obs.: Quando se for mudar uma propriedade comum a vários objetos, é mais rápido fazer a mudança em “linha de produção”, ou seja quando for selecionado o próximo objeto, o VB mantém o cursor na propriedade que foi recentemente alterada.

. Executar o programa (mesmo sem ter feito o código) para verificar se a tela está de acordo com o esperado.

Passos recomendados no item 3:

Obs. : Antes da execução dos passos abaixo supõe-se que o programador já elaborou a lógica do programa que vai codificar em VB ( por  meio de uma das técnicas de diagramação - fluxograma, portugol, etc.), e que já sabe exatamente o que o programa vai fazer.

. Abrir a janela de código, dando um duplo click no objeto escolhido;

. Selecionar o protótipo de procedure que interessa.

. Digitar o programa em VB.

. Testar o programa para verificar se o resultado está de acordo com o esperado.

EXERCÍCIO: Elaborar a tela e fazer um programa de acordo com o descrito abaixo:

Quando clicarmos o botão “click-me”, no quadro de texto surgirá a mensagem “Bem vindos ao Visual Basic”.

Na figura acima visualizamos a janela de código do VB contendo:

-Lista “drop down”, mostrando os nomes dos objetos que compõe a aplicação;

-Lista “drop down”, contendo as procedures possíveis para o objeto selecionado.  Se a abrirmos, poderemos ver e selecionar qualquer uma das procedures que tratam os eventos possíveis para o objeto selecionado.

-”Procedure Sub” que trata o evento Click para o objeto “cmdclickme”. Note que o protótipo das “Subs” já vem pronto, bastando apenas incluirmos o código que será executado. Note também que a forma das “Subs” é sempre:

  Private Sub nomedoobjeto_evento( ...)

          PROGRAMA EM VB

  End Sub 

Continuando o exercício, faremos com que o nosso programa execute de acordo com o seguinte procedimento:

a) O Caption do botão inicia com “click-me” e a caixa de texto inicialmente não contém nenhum texto ( sua propriedade “text” contém espaços em branco ou o carácter nulo).

b) Quando se dá um click no botão, surge a mensagem “Bem vindos ao Visual Basic” na caixa de texto.

c) Se fizermos alguma alteração nos dizeres da caixa de texto, o caption do botão torna-se “Restaura”.

d) Clicando-se novamente o botão, o text da caixa de texto é restaurado para “Bem vindos ao Visual Basic” e o caption do botão volta a ser “click-me”.

O CÓDIGO DO PROGRAMA AGORA, PASSARÁ A SER:

Private Sub Cmdclickme_click( )

        Txtmensagem.text=“Bem vindos ao Visual Basic”

        Cmdclickme.caption=“click-me”

End Sub

Private Sub Txtmensagem_change( ) 

        cmdclickme.caption=“Restaura”

End Sub

Neste programa vamos apontar um detalhe que mostra a diferença gritante que existe entre a programação tradicional e a programação orientada a eventos.

Na programação tradicional não haveria problema algum se invertêssemos as instruções da Sub Cmdclickme_Click( ), pois como as instruções são executadas sequencialmente, não deve haver problemas na ordem de execução das instruções: 

Private Sub Cmdclickme_click( )

        Txtmensagem.text=“Bem vindos ao Visual Basic”

        Cmdclickme.caption=“click-me”

End Sub

Vamos invertê-las deixando-as da seguinte maneira:

Private Sub Cmdclickme_click( )

      Cmdclickme.caption=“click-me”

      Txtmensagem.text=“Bem vindos ao Visual Basic”

End Sub

Execute o programa agora e verifique que já não funciona mais a contento. Explique porquê!
Você acabou de verificar que o programador que trabalha com programação orientada a eventos deve se preocupar com os eventos gerados pelo operador e com os eventos oriundos da própria aplicação que está sendo executada.

Com a programação orientada a eventos, não existe uma sequência pré estabelecida como na programação tradicional. O operador pode escolher uma ação ou outra dependendo de sua vontade, fazendo com que ocorram eventos na aplicação, sem nenhuma sequência previsível.

Além dos eventos gerados diretamente pelo operador, a própria aplicação gera eventos automaticamente.

Quando foi executada a instrução  Txtmensagem.text=“Bem vindos ao Visual Basic”, a mudança na propriedade text, provocou o evento “Change” na caixa de texto Txtmensagem, executando novamente a Sub Txtmensagem.Change, mudando novamente o caption do botão para “restaura”.

Portanto, tomar cuidado com os eventos gerados automaticamente pela própria aplicação, como no exemplo abaixo:

                         [image: image6.png]



Private Sub TxtErro_Change()

   TxtErro.Text = TxtErro.Text + "x"

End Sub

Execute o programa acima, digite algo na caixa de texto e explique o que aconteceu!!!

EXERCÍCIO: Fazer um programa “E C O” segundo o layout a seguir:

Queremos que o programa faça “ecoar” no retângulo inferior, tudo que se digita no retângulo superior.

O retângulo superior é uma caixa de texto (textbox) e o inferior é um label com a propriedade BorderStyle=“fixed single”para ficar parecido com caixa de texto. Como você sabe, labels não permitem que o operador digite nada neles, apesar de podermos escrever neles por meio do nosso programa. 

Para que o programa funcione da maneira descrita, digite o código abaixo, usando o protótipo da “Private Sub Txtdigita_Keypress”:

Private Sub Txtdigita_KeyPress (keyascii As Integer)

      Dim tecla As String

      tecla = Chr$(keyascii)

      lbleco.Caption = lbleco.Caption + tecla

End Sub

Quando ocorre o evento Keypress no objeto Txtdigita, é executado o código contido na Sub Txtdigita_Keypress(. . .). O argumento keyascii é uma variável de valor inteiro que contém o valor numérico em decimal do carácter ASCII da tecla pressionada. Esse valor é convertido para carácter pela função Chr$, atribuído para a variável “tecla” e concatenado no Caption de “lbleco”.

EXERCÍCIO: Alterar o programa acima para ecoar em lbleco apenas valores não numéricos.
Obs.: Comando IF: forma geral:

if condição_verdadeira then

     instrução1

     instrução2

         ...

else

     instruçãok1

     instruçãok2

        ....

end if

Exemplo:  if tecla < “0” or tecla > “9” then lbleco.caption=lbleco.caption + tecla 

REFAZER o exercício para ecoar agora somente valores numéricos.

EXERCÍCIO: Fazer um programa “calculadora” segundo o layout a seguir:

Código associado à janela acima: 

Private Sub CMDIGUAL_Click ()

     Dim N1 As Single

     Dim N2 As Single

     Dim RESULT As Single

     N1 = Val(TXTN1.Text)   ‘ isto poderia ser feito de maneira mais simples!!!!!

     N2 = Val(TXTN2.Text)

     RESULT = N1 + N2

     TXTN3.Text = RESULT

End Sub

Private Sub TXTN1_KeyPress (KEYASCII As Integer)

      Dim TECLA As String

      TECLA = Chr$(KEYASCII)

      If (TECLA < "0" Or TECLA > "9") And TECLA <> "." Then

           Beep

           KEYASCII = 0

      End If

End Sub

EXERCÍCIOS:

- Utilizar a propriedade TabStop para o cursor não ir à caixa de texto do TxtN3;

- Definir a propriedade “default”do botão “CmdIgual” para TRUE;

- Utilizar o “help on line”  (tecla F1);

- Substituir a caixa de texto TxtN3 por um label (fazer o “borderstyle” fixed single);

- Incluir o botão “Limpa”, que, ao ser pressionado, limpa todos os números da calculadora e posiciona o cursor na primeira caixa de texto (txtn1). Para posicionar o cursor em “txtn1”, utilizar o método SetFocus ( ex. Txtn1.SetFocus);

- Utilizar o recurso de TabIndex, para mudar a ordem da seqüência do foco, ao ser pressionada a tecla “Tab”.;

- Copiar o código de TxtN1 para TxtN2 usando o recurso de cópia ( CTRL+C, CTRL+V);

- Formatar a saída do resultado, com dois decimais  : FORMAT(RESULT, “$##,###.00”);

- Incluir uma “Caixa de Mensagem” (MSGBOX) para dar uma mensagem de erro quando for digitado um valor não numérico, ex.: MSGBOX “TECLA INVÁLIDA”, 48, “ A V I S O “ (utilize o HELP com a tecla F1 e verifique as outras opções;

-Modificar as propriedades abaixo no formulário e verificar seus efeitos no form.

MAXBUTTON, MINBUTTON, CONTROLBOX, BACKCOLOR.

-Modificar as propriedades abaixo em um dos botões e verificar seus efeitos.

ENABLED, BACKCOLOR, MOUSEPOINTER, VISIBLE.

Obs.: Para verificar a propriedade BackColor, alterar a propriedade Style, para Graphical.

-Modificar o programa para que aceite nas caixas de texto a tecla BackSpace (KeyAscii <>8)

-Modificar o programa para que só permita a digitação de um ponto decimal.

Obs.: a função InStr( Text1.Text,”.”)  retorna 0 se não achou o “.” ou retorna a posição onde encontrou o “.”. Verifique o Help on line.

-Estando o foco em TxtN1, ao pressionar “Enter”, o foco deve ir para TxtN2.

-Coloque um ícone em sua calculadora.

- REFAZER O PROJETO DA CALCULADORA, PARA FAZER AS QUATRO OPERAÇÕES.

Sugestão: Copiar o código da função que trata o evento CLICK para os outros botões, usando as teclas de atalho do WINDOWS (para copiar: CTRL+C, para colar: CTRL+V).

Em seguida, alterar o operador aritmético, para fazer a operação aritmética correspondente a cada botão.

-Ao pressionar uma das teclas + - * /, a operação correspondente é realizada. Exemplo:

    if Keyascii = 43 then CmdMais_Click

Obs.: Cuidado com a operação de divisão... Não existe divisão por zero... Como contornar esse problema?

-Alterar o programa para que não haja a possibilidade de divisão por zero, pois essa divisão é impossível. Nesse caso, o programa deve:

- exibir uma caixa de mensagem (MsgBox), avisando ao usuário sobre o erro;

- posicionar o cursor no local do denominador e deixar os números selecionados para que o valor seja corrigido. Para isso, utilizar:

TxtN2.SetFocus

TxtN2.SelStart=0

TxtN2.SelLength=Len(TxtN2.Text)

-Incluir um botão para calcular a raiz quadrada do valor que está em TxtN1. Tratar a raiz de valores negativos, da mesma maneira que tratou a divisão por zero. Este botão dever ser colorido e possuir o símbolo da raiz quadrada.

Obs.: Cuidado com as conversões automáticas do VB.

-Incluir o botão Fim que ao ser pressionado encerra o processamento do programa (instrução End).

-Se o usuário tentar fechar o programa através da caixa de arquivo, de Alt+F4 ou através do botão fechar, o programa pergunta ao usuário “ Quer mesmo sair?”. Para isso, utilizar o evento Form_Unload:

Private Sub Form_Unload(Cancel as Integer)

    If MsgBox(“Quer mesmo sair?”, VbYesNo)=VbYes then

         Unload Form1

   Else

        Cancel = True

   EndIf

End Sub
OBS.: 1 - Você pode mudar o “GRID” do formulário, se achar que precisa de maior ou de menor precisão para movimentar ou posicionar os objetos na tela. Para isso, abra o item de menu “Tools”, selecione “Options”, escolha a folha “General” e altere os itens “Grid Width” e/ou “Grid Height”. Experimente mudá-los para 30 twips e veja o que acontece. 

2 - Note que há várias guias onde são agrupadas as opções da interface visual do VB. As modificações que forem feitas, serão gravadas e continuarão valendo nas próximas seções que se utilizar o VB.

Experimente desligar a opção “Show Tool Tips”, dando um click na “CheckBox” correspondente a essa opção. Note que não são mais apresentadas as “dicas” que apareciam na interface do VB.

3 - Qualquer esclarecimento sobre as opções, pode ser obtido por meio do Help.

TIPOS DE VARIÁVEIS USADAS EM VB

Variável: É a maneira pela qual se manipulam dados em programação. Quando se atribuem nomes às variáveis usadas em um programa, deve-se escolher um nome que tenha relação com o conteúdo que a variável vai assumir. Assim, uma variável que vai armazenar o salário dos funcionários em um sistema de folha de pagamento, seria melhor atribuir-lhe o nome “SALARIO”, do que chamá-la de “X” ou outro nome qualquer, que não tenha significado.

Em VB é possível trabalharmos com os tipos de variáveis abaixo:

TIPO

CARÁCTER 

NUM. DE BYTES
VALORES POSSÍVEIS

INTEGER

%


2

     -32768    A   32767

LONG


&


4

-2147483648 A 2147483647

SINGLE

!


4

-3.37 x 1038      A   3.37 x 1038
DOUBLE

#


8

-1.67 x 10308   A    1.67 x 10308
CURRENCY

@


8

-9.22 x 1014      A    9.22 x 1014
STRING

$


1 POR CARÁCTER

BYTE





1

0    A    255 

BOOLEAN




2

TRUE ou FALSE

DATE





8

1/1/100   A   12/31/9999

OBJECT




4

Figuras e objetos OLE.

VARIANT




VARIA, dependendo do dado atribuído.

Obs.: A declaração da variável pode ser explícita, ou seja, declara-se a variável e o seu tipo.

Exemplo: dim salario as double
Ou pode ser declaração implícita, onde o aparecimento do nome da variável e o carácter do tipo desejado, é suficiente para que se possa usar a variável.

Exemplo:  salario# = Val( Txtsalario.text)

No exemplo acima, o VB automaticamente cria uma variável do tipo double de nome salario# e atribui a ela o valor numérico que estava na caixa de texto de nome Txtsalario.

A opção pelo programador por uma destas modalidades de declaração, deve ser feita levando-se em conta as vantagens e desvantagens de cada uma. Se o programador optar pela obrigatoriedade da declaração de variáveis, deve incluir manualmente na área “General” do formulário, a cláusula “Option Explicit”.

O VB faz isso automaticamente ao abrirmos um novo projeto, se selecionarmos o item Tools/Options/Editor no menu do VB.


Em VB a declaração de variáveis pode ser feita em 4 lugares:





USO DO CONTROLE “TIMER” (TEMPORIZADOR)
Exercício: fazer um programa “CRONÔMETRO” segundo o layout abaixo:

Código associado à janela acima:

Dim INICIO As Variant, FIM as variant    ‘variáveis declaradas na área GENERAL.

Private Sub Timer1_Timer () 

                       ‘Temporizador com INTERVAL=500 : a cada meio segundo,

                       ‘                            executa o código dentro desta procedure.

     Lblhoratual.Caption = Time$

End Sub

Private Sub Cmdiniciar_Click ()

      Lblhorainiciou.Caption = Time$

      INICIO = Now

      Lblparou.Caption = “”

      Lbltdecorrido.Caption = “”

End Sub

Private Sub Cmdparar_Click ()

     Lblparou.Caption = Time$

     FIM = Now

     LblTdecorrido.Caption = Format(FIM - INICIO, "HH:MM:SS")

End Sub

Private Sub Form_Click ()

    Dim Hoje   ' Declaração de variável

    Hoje = Now ' Pega hora e data atual.

    MsgBox "Hoje é ”  & Format(Hoje, "dddd, mmmm dd, yyyy") & "."

End Sub

Exercício: Ao executar o programa, o botão INICIAR estará habilitado e estará desabilitado o botão PARAR. Ao receber o Click, o botão INICIAR ficará desabilitado e habilitará o botão PARAR. Quando o usuário clicar no botão PARAR, o botão INICIAR será habilitado e desabilitará o botão PARAR.

Obs. :

- O objeto Temporizador (Timer) não aparece no formulário em tempo de execução.

- A propriedade “interval”do temporizador é especificada em milissegundos, variando de 1 a 65.535 ( pouco além de 1 minuto). Não esqueça de especificar a essa propriedade um valor diferente de zero, ou a aplicação não funcionará

-A função TIME$ retorna a hora do sistema em caracteres.

 -A função NOW é necessária para a subtração. Ela retorna uma variável do tipo variant, com um  valor numérico de dupla precisão (double) contendo a data e hora do sistema (vide o help on line - F1).

-O objeto “Timer” não está diretamente relacionado a aplicações que envolvam controle de tempo, tais como segundos , minutos, etc. Acompanhe o exercício a seguir.

O temporizador é bastante utilizado em aplicações que envolvem animação de figuras e objetos. Vamos ver alguns exemplos:

Utilizar as ferramentas “picture” e “shapes” para criar a tela abaixo:

 Option Explicit

Private Sub Timer1_Timer ()

    picture2.Visible = Not picture2.Visible

    If  picture2.Visible Then Beep

End Sub

Execute o programa na forma que está. Em seguida, pare o programa, arraste a picture2 (marreta levantada) para ficar  escondendo a picture1(marreta que bate) e execute o programa para visualizar o efeito de animação. Muitos programas usam esse recurso.

Para obter um efeito melhor, faça a propriedade “borderstyle” das pictures  = “none”

Poderíamos “encaixar” no objeto picture, quaisquer tipos de figuras, tais como ícones, desenhos feitos pelo Paintbrush, etc. Se especificarmos a propriedade “autosize” da picture = “true”, esta irá adaptar-se ao tamanho da figura que irá conter.

Vamos incrementar mais um pouco nosso trabalho de animação, incluindo no formulário o controle “Slider Control”. Esse controle faz parte de um conjunto de ferramentas chamado “Microsoft Windows Common Controls” e só é suportado pelo pacote do Windows 95 ou superior.

Caso o “Slider Control” não esteja na “Toolbox”, você deverá incluí-lo, juntamente com outros controles que fazem parte do conjunto. Para isso, basta abrir o item do menu “Project/Components”, escolher a guia “Controls” e selecionar o “Microsoft Windows Common Controls”. Após dar um click com o mouse na check-box desse item, marcando-o, pressione o botão “Ok”, que vários componentes, inclusive o Slider Control, serão acrescidos à sua ToolBox.. 

Se o “Microsoft Windows Common Controls” não estiver disponível, podemos incluí-lo na lista de controles disponíveis. Esse conjunto de controles está no arquivo “COMCTL32.OCX”, que está no diretório “C:\WINDOWS\SYSTEM”.

Obs.:

Podemos incluir novos controles em nossa “Toolbox” por intermédio da janela “Components”.  Para isso, devemos dar um click no botão “Browse” e localizar o local onde está o controle que queremos incluir. Deverão ser arquivos com extensão “.OCX”.

Em nosso caso vamos selecionar e incluir o arquivo “COMCTL32.OCX”, caso ele ainda não faça parte de nossa lista de ferramentas disponíveis, na janela “Components”.

Estando o “Slider Control” em nossa “Toolbox”, vamos incluí-lo em nossa aplicação que deverá ficar mais ou menos com a seguinte aparência:

O “Slider Control” pode ser utilizado para devolver valores dentro de uma faixa estabelecida em suas propriedades “Max” e “Min”. Se por exemplo especificarmos Max com o valor 10 e Min com o valor 1, os valores devolvidos pela propriedade “Value”, estarão na faixa de 1 a 10. 

Vamos agora utilizar o valor retornado pelo “Slider Control” para modificar a nosso critério a propriedade “Interval” do temporizador “Timer1” de nossa aplicação.

Como já dissemos, o valor que usaremos do “Slider Control” vem da propriedade “Value”. Assim, se usarmos a procedure “Change” do “Slider Control”,  sempre que fizermos nele alguma alteração, por meio de mouse , das setas de direção ou pelas teclas PgUp/Pgdn, iremos alterar o “Interval” de nosso temporizador, mudando a velocidade de nossa animação.

Inclua na aplicação o código abaixo:

Private Sub Slider1_Change()

     Timer1.Interval = Slider1.Value * 50

End Sub  

Execute o programa e verifique que a animação agora pode ser controlada movimentando-se o cursor do Slider.

Observe que neste caso, quando se aumenta o valor do Slider ( movimenta-se o cursor para a direita), a animação fica mais lenta. Isso deve-se ao fato de que quanto maior o “Interval” do temporizador,  maior é o período de tempo entre duas execuções. Para deixarmos a velocidade da animação diretamente proporcional ao incremento do Slider, podemos refazer o código de acordo com o seguinte programa completo:

Private Sub Slider1_Change()

    Timer1.Interval = 500 / Slider1.Value

End Sub

Private Sub Timer1_Timer()

    Picture2.Visible = Not Picture2.Visible

End Sub

Atenção:... Cuidado com a propriedade “Min” do Slider, que, neste caso, não pode ser 0. Como estamos calculando o Interval de nosso temporizador através da expressão: “Timer1.Interval = 500 / Slider.Value”, se o valor de “Min” for 0 e trouxermos a régua para o valor mínimo, haverá erro de processamento, pois será feita uma tentativa de divisão por zero!!!

Para corrigir esse problema, atribua o valor 1 à propriedade Min do Slider.

O “Slider Control” pode ser utilizado com bastante eficiência em aplicações em que é necessário o retorno ou seleção de  valores inteiros sem precisarmos digitá-los pelo teclado. 

Para explorarmos outras ferramentas e recursos do VB, vamos desenvolver a aplicação “DESPERTADOR” a seguir:

Obs.: 

-Botões de opção (ou botões de rádio) são usados quando se deseja selecionar uma opção dentre várias. Quando se dá um click em um deles, se havia um outro que estava selecionado, este será automaticamente desativado. Pode-se ter mais de um conjunto de botões de opção, desde que dentro de um “frame”.

-Note que utilizamos aqui o evento “Form Load”. Esse evento ocorre no momento em que a aplicação em VB é executada. O “Form Load” é utilizado para inicializar variáveis, setar propriedades e condições iniciais necessárias ao início da aplicação.

Código associado à janela do DESPERTADOR acima:

Option Explicit

Private Sub Form_Load ()

     optpara = True ‘    COMEÇA COM O DESPERTADOR DESLIGADO!

End Sub

Private Sub Timer1_Timer ()

     Lblhora.Caption = Time$

     If Time$ >= txtdesperta.Text  And  optliga Then

         Beep

     End If

End Sub

Private Sub Txtdesperta_KeyPress (keyascii As Integer) 

                       ‘*************CUIDADO!!! O evento é Keypress() e não Change().

     Dim tecla As String

     tecla = Chr$(keyascii)

     If (tecla < "0" Or tecla > "9") And tecla <> ":" Then

         Beep

         keyascii = 0

         MsgBox "Digite apenas valores de '0' a '9' ou o carácter ':' ", 48, "A T E N Ç Ã O"

     End If

End Sub

Vamos incrementar um pouco mais nosso despertador, para ficar com o seguinte aspecto:

Incluir um botão (Cmdmensagem) que,  quando receber um click, abre uma “InputBox” para que se digite uma mensagem qualquer.

Inclua um label de nome LblMensagem, com a propriedade visible = false. O Caption desse label irá receber a mensagem digitada na InputBox.

Quando despertar, aparecerá a mensagem que digitamos na “InputBox”, como por exemplo:

Obs.:

O Caption do botão “MENSAGEM” foi escrito “&MENSAGEM” para que se possa acessá-lo diretamente por meio das teclas “ALT + M”.

Se definirmos a propriedade “DEFAULT” deste botão para “true”, ele aparecerá com a borda mais escura, podendo ser acessado diretamente pela tecla “Enter”.  Em uma aplicação, é possível ter apenas um botão com a propriedade “DEFAULT=true”.

A propriedade WindowState estabelece o valor 0 para janela normal, 1 para janela minimizada e 2 para janela maximizada. Assim, se nosso despertador estiver minimizado ( será apenas um ícone), ao despertar, será maximizado, ficará apitando, piscando a mensagem e ocupará a tela toda.

Verifique, após alterar o programa para ficar como abaixo:

Option Explicit

Private Sub Cmdmensagem_Click ()

      lblmensagem.Caption = InputBox("Entre a Mensagem...")

End Sub

Private Sub Form_Load ()

     optpara = True    ‘ .....................DEIXA ALARME DESLIGADO.

     lblmensagem.Visible = False  ‘..FAZ O LABEL DE MENSAGEM INVISÍVEL.

End Sub

Private Sub Timer1_Timer ()

     Lblhora.Caption = Time$

     If Time$ >= txtdesperta.Text And optliga Then

            Beep

            lblmensagem.Visible = Not lblmensagem.Visible   ‘SE ESTIVER VISÍVEL FICA 

                                                                                             ‘INVISÍVEL E VICE VERSA.

            If formdesperta.WindowState = 1 Then       ‘SE ESTIVER MINIMIZADO,

                    formdesperta.WindowState = 2             ‘QUANDO DESPERTAR

            End If                                                             ‘ SERÁ MAXIMIZADO.

     End If

End Sub

Private Sub TXTDESPERTA_KeyPress (keyascii As Integer)

     Dim tecla As String

     tecla = Chr$(keyascii)

     If (tecla < "0" Or tecla > "9") And tecla <> ":" Then

         Beep

         keyascii = 0

         MsgBox "Digite apenas valores de '0' a '9' ou o carácter ':' ", 38, "A T E N Ç Ã O"

     End If

End Sub

Salvar o arquivo executável deste  programa em seu disco.

Vamos fazer mais um exercício de animação, usando o temporizador e uma malha “for”, que é usada na programação tradicional, assim como as malhas usadas nas várias linguagens, tais como malhas “while”, “do - until”, “perform”, etc.

Verificaremos que em uma linguagem orientada a eventos, nem sempre é vantagem a utilização de uma malha, podendo eventualmente, ser substituída pelo uso de um temporizador.

Faremos um programa para o lançamento de um foguete, que enquanto vai subindo para o espaço, vai deslocando-se para a direita e diminuindo de tamanho, dando a idéia de ir distanciando-se.

Projete a tela de acordo com o layout abaixo:

Para abrigar a figura do foguete utilizaremos o controle “image”. Lembre-se que no exercício de animação dos martelos, nós utilizamos o controle “picture”. Os dois controles são bastante parecidos, porém as propriedades de cada um deles são distintas. Por exemplo, “picture” tem a propriedade “autosize” que faz com que o tamanho do controle “picture” ajuste-se à figura que carregarmos nele. O controle “image” tem a propriedade “stretch” que permite alterarmos a largura e altura da figura nele contida.  Esta é a propriedade que iremos usar para que o foguete vá diminuindo de tamanho.

O controle “picture” tem mais propriedades que “image”. Por outro lado, o controle “image” é acessado mais rapidamente.

Após desenhar a tela abaixo, atribuir “true”para a propriedade “stretch” da image  de NAME=“figfoguete” ;

Para trazer o desenho do foguete, abra a propriedade “picture”e escolha o arquivo C:\arquiv~1\devstu~1\vb\graphics\icons\industry\rocket.ico.

OBS.: AS UNIDADES DE DISTÂNCIA DOS OBJETOS ÀS BORDAS DO FORMULÁRIO PODEM SER EXPRESSAS EM “TWIPS”. ESSA É UMA UNIDADE DE MEDIDA QUE CORRESPONDE A APROXIMADAMENTE 1/567 Cm. (ou 0,02mm)

A princípio, vamos fazer o foguete subir através de uma malha:

Private Sub cmdispara_Click ()

Dim i As Integer

     For i = 1 To 500                          ‘ EXECUTA A MALHA 500 VEZES

           figfoguete.Top = figfoguete.Top - 3     ‘UNIDADE EM TWIPS

           figfoguete.Left = figfoguete.Left + 1

           figfoguete.Width = figfoguete.Width - 1

           figfoguete.Height = figfoguete.Height - 1

      Next i

End Sub

Private Sub cmdpara_Click ()

     end  ‘   Pára o programa

End Sub

Execute o programa, dando um click no botão “Dispara” e em seguida tente pará-lo , clicando o botão “PÁRA”, para ver o que acontece...  (Explique por que o foguete não parou imediatemente ou por que nada mais funciona por um certo intervalo de tempo...).

O que acontece é que quando você dá um click no botão “cmdispara”, o programa começa  a executar uma malha de 500 iterações.

Como provavelmente sua máquina tem apenas um processador, nada mais funcionará enquanto a malha que está sendo processada não terminar sua execuçaõ e “soltar” o processador.

Por esse motivo, deve-se pensar muito bem antes de usar malhas em programas com orientação a eventos. Muito provavelmente, o usuário do programa vai pensar que o computador travou ou então quebrou ! . Talvez pense até que seu programa não é bom (que absurdo...).

A solução para o problema acima é muito simples. Basta trocar a “malha for” por um temporizador que este fará o trabalho sem “prender” o processador. 

Deixe como “false”a propriedade “enabled” do temporizador e faça o “interval” = 100 milissegundos, por exemplo.

Código modificado, associado à tela acima:

Private Sub cmdispara_Click ()

     timer1.Enabled = True   ‘    LIGA O TEMPORIZADOR

End Sub

Private Sub cmdpara_Click ()

     timer1.Enabled = False   ‘   DESLIGA O TEMPORIZADOR

End Sub

Private Sub Timer1_Timer ()

           figfoguete.Top = figfoguete.Top - 10     ‘UNIDADE EM TWIPS

           figfoguete.Left = figfoguete.Left + 5

           figfoguete.Width = figfoguete.Width - 2

           figfoguete.Height = figfoguete.Height - 2

End Sub

Como fizemos o “interval”=100, a cada 100 milissegundos o código dentro da Sub Timer1_Timer()  será executada, causando um efeito parecido com o da malha que estávamos usando.

Mude o valor do “interval” para valores diferentes e visualize o resultado.

Exercício: Vamos incrementar um pouco mais nosso  programa do foguete, para que o mesmo ao subir, choque-se com um avião que vem voando em sua direção. Para que isso não lhe cause choques traumáticos, suponha que não há passageiros no avião e nem no foguete.

O procedimento pode ser algo assim:

O foguete vai subindo e diminuindo de tamanho;

O avião vem se aproximando e aumentando de tamanho;

Quando houver uma certa distância mínima entre os dois, eles param de se movimentar, desaparecem,  e surge a imagem de uma explosão. 

A tela do projeto pode ser algo como abaixo:

Obs.: O ícone do avião está no mesmo diretório do foguete, e tem o nome de “Plane.ico”.

Analise o código de programa abaixo:

Option Explicit

Dim i As Integer

Private Sub CmdDispara_Click()

Timer1.Enabled = True

End Sub

Private Sub CmdPara_Click()

Timer1.Enabled = False

End Sub

Private Sub Timer1_Timer()

  If i = 1 Then

     figexplosao.Visible = Not figexplosao.Visible

  Else

     If Abs(figfoguete.Top - figaviao.Top) < 300 And Abs(figfoguete.Left - figaviao.Left) < 300 Then

        i = 1   ‘No próximo ciclo do Timer, a figexplosao ficará visível.

        figfoguete.visible=false    ‘No próximo ciclo, fará o foguete o aviao invisivel. 

        figaviao.visible=false

     Else

        figfoguete.Top = figfoguete.Top - 10

        figaviao.Top = figaviao.Top + 10

        figfoguete.Left = figfoguete.Left + 7

        figaviao.Left = figaviao.Left + 7

        figfoguete.Width = figfoguete.Width - 2

        figaviao.Width = figaviao.Width + 1

        figfoguete.Height = figfoguete.Height - 2

        figaviao.Height = figaviao.Height + 1

      End If

  End If

End Sub

Exercícios:

-Incluir o botão “Fim”, com o mesmo tratamento que demos ao botão “Fim” da calculadora.

-Assim que ocorrer o choque, a figura da explosão vai diminuindo até desaparecer.

-Assim que ocorrer o choque, um sobrevivente salta de paraquedas e vai descendo até desaparecer.

-Incluir o botão “Acelera...”, que a cada click que recebe, acelera a velocidade horizontal do avião em 1 twip a cada ciclo do temporizador.

Obs.: Salve esse programa em seu disco de trabalho que posteriormente iremos incrementá-lo, quando virmos “Criação de Números Aleatórios”.
Novo ´projeto:

Vamos fazer mais um exercício que dê idéia de animação e mostraremos como se faz um ícone animado ( o ícone continuará animado na barra de tarefas, mesmo se for minimizada a janela do programa).

Fazer a tela do programa “LUAR”,  de acordo com o layout abaixo:

Obs.:

- Utilizar o objeto “image” com “stretch”= true para inserir o desenho do planeta terra;

- Inserir na propriedade “picture”o icone de endereço :

“C:\arquiv~1\devstu~1\vb\graphics\icons\elements\earth.ico”

LEIA COM ATENÇÃO AS INSTRUÇÕES ABAIXO, MAS ANTES DE EXECUTÁ-LAS, LEIA O QUE ESTÁ ESCRITO MAIS ABAIXO, EM (***OBS.:***) E EM SEGUIDA, DECIDA QUAL MANEIRA VOCÊ QUER FAZER O TRABALHO: A MAIS TRABALHOSA OU A MAIS FÁCIL E RÁPIDA.

- Utilizar o objetos “picture” para inserir os desenhos de luas;’

- Fazer a propriedade “autosize”=true e “visible”=false;

ATENÇÃO:  Se você criar mais de um controle com o mesmo nome, o VB irá considerar que você quer criar uma “MATRIZ DE CONTROLES”, ou um “CONTROL ARRAY”. No caso deste exercício, isto é verdade. Como irão ter o mesmo nome, cada um dos controles “picture” que irá conter o ícone de um tipo de lua, será acessado por seu nome e um índice.

Você poderia ir acrescentando os controles “picture” da maneira tradicional e depois de inserir os 8 controles “picture”, fazer o “name” de cada um deles = pict1, por exemplo.

Nesse momento, o VB irá perguntar se você quer criar um “Control Array”, e você deve nesse caso responder afirmativamente.

(*** OBS.:***) Uma outra maneira (mais rápida) de fazer isso, é inserir na tela apenas o primeiro controle , adequar todas as propriedades desejadas, tais como NAME=pict1, AUTOSIZE=true, BORDERSTYLE=none, VISIBLE=false.

Em seguida, utilizamos o recurso de cópia do WINDOWS, por meio das teclas Ctrl C (para copiar) e Ctrl V ( para colar). Para isso, basta selecionar a picture que foi configurada e pressionar Ctrl C. Daí, damos um click no formulário para selecioná-lo, e pressionamos Ctrl V. Em seguida, movimentamos o controle copiado para o local desejado, e vamos digitando agora apenas Ctrl V para ir colando as novas cópias do controle, que possui todas as propriedades da cópia original.

- Inserir na propriedade “picture” os ícones do endereço:

                                C:\arquiv~1\devstu~1\vb\graphics\\icons\elements\moonn.ico;

- O temporizador deve ter a propriedade “enabled”= false e o “interval” pode ser 500. 

- Se seu monitor for colorido e você fizer o fundo da tela preto, o efeito será muito bonito. Faça preta também a propriedade Backcolor das pictures que contém os ícones de luas.

Tela do programa “LUAR”:

O código associado à tela acima é o seguinte:

Dim n As Integer    ‘A variável n é global a este formulário,

                               ‘ou seja é declarada na seção “(general)”.

Private Sub cmdpara_Click ()

     timer1.Enabled = False  ‘desabilita o temporizador

End Sub

Private Sub cmdvai_Click ()

     n = 0

     pict1(0).Visible = True

     timer1.Enabled = True  ‘habilita o temporizador

End Sub

Private Sub Form_Load ()  ‘Poderia já estar setado False em tempo de 

                                          ‘  projeto.

     timer1.Enabled = False

End Sub

Private Sub Timer1_Timer ()

      Dim m As Integer

      m = n                                ‘esta lógica faz com que a

      n = n + 1                           ‘cada vez que o temporizador 

      If n > 7 Then n = 0            ‘é habilitado, a pict1(n) fica invisível

      pict1(m).Visible = False     ‘e a pict1(n+1) fica visível.

      pict1(n).Visible = True      ‘ANALISE-A P/ VERIFICAR...

     form1.icon=pict1(n).picture  ‘quando minimizado, a lua ficará girando

End Sub

Exercício: Quando ocorrer o click no botão “Pára”, a última lua que estava visível deve ficar invisível. Altere o programa para que isso aconteça.

Obs. Caso você tenha inadvertidamente inserido alguma figura (ou algum ícone) na propriedade “Picture”do formulário e deseja eliminá-la, adote o seguinte procedimento: 

- Selecione o formulário;

- Selecione a propriedade Picture;

- Posicione o cursor sobre a palavra “Icon” ou “Bitmap” e dê um click com o mouse;

- Pressione a tecla “Del”.

*** Minimize a janela do programa “ L U A R” e verifique o ícone girando na barra de tarefas. Note que o programa “Luar” continua sendo executado, mesmo se você começar a trabalhar em outro aplicativo (como o Word, por exemplo).

Nosso próximo trabalho tem por objetivo a apresentação e uso dos seguintes recursos:

- Utilização de botões de opção dentro de uma “Frame”; 

- Uso da função “date”;

- Finalização de um programa por meio do “End”;

- Manipulação de soma de strings;

- Uso do método “Print”.

EXERCÍCIO: Desenvolva a interface (tela) do programa “JOGO” abaixo:

[image: image18.png]w Jogo hora:03:29:56  [M[=]|
1 fogador nivel de diiculdade =5

Opeo
& 1jogador Iniciar
€ 2iogadares

Fim

Nivelde Dificuidade (129)
Quintaeira, Janeio 02,1987





O programa deve fazer o seguinte:

No Caption do formulário deve aparecer a hora atual, sendo atualizada a cada meio segundo. Para isso, incluir o controle “Timer” com a instrução:

                 Form1.Caption = "Jogo                    hora:" + Time$

A cada click do mouse dado no botão “Iniciar”, devem ser impressas no formulário as opções selecionadas (Se um ou dois jogadores e o nível de dificuldade). Para isso, programar o botão “Iniciar” para que no evento Click, execute as seguintes instruções:

Dim X As String

If Option1.Value = True Then X = X + "1 jogador"

If Option2.Value = True Then X = X + "2 jogadores"

X = X + " nível de dificuldade =" + Text1.Text

Print X

No rodapé do formulário deve aparecer a data do sistema. Para isso, incluir também no código da Private Sub Timer1_Timer(), a instrução:

    Label1.Caption = Format(Date, "dddd, mmmm dd,yyyy")

Ao ser dado um click no botão “Fim”, o programa deve exibir uma caixa de mensagem com a mensagem “Fim do jogo”. Para isso, programar o botão “Fim”com o seguinte código: 

Private Sub Command2_Click()

    MsgBox "Fim do jogo"

    End

End Sub

MANUSEIO DE BANCO DE DADOS

O VB tem em seu conjunto o aplicativo “Visual Data Manager”, que permite a criação e manipulação de vários tipos de bancos de dados, tais como “ACCESS”, dBase, FoxPro, etc.

Vamos utilizá-lo para criar um banco de dados compatível com o ACCESS, que posteriormente iremos acessar por meio do VB.

Para isso, abra o item de menu “Add Ins”  e dê um click no item “Visual Data Manager”.

Quando abrir a janela do “VisData”, dar um click no item “File”, e escolha “New/Microsoft Acess/Version 7.0”.

Ao surgir a janela para salvar arquivo, digitar “A:REVENDA” no campo “Nome do Arquivo”, e dar um click em “Salvar”.

Quando surgir a janela “Database Window”, dar um click, com o botão direito do mouse, no item “Properties”, e em seguida, dar um click em “New Table”. 

Ao sugir a tela “Table Structure”, digitar no campo “Table Name”, o nome de nossa tabela: “Veiculos”. Em seguida, dar um click no botão “Add Field”.

Ao abrir a janela “Add Field”, preencher os campos Name, Type e Size, de acordo com as especificações abaixo, para criar a estrutura do arquivo.

Após o preenchimento dos dados de cada campo, clicar o botão “Ok“ para ir ao próximo campo.

NOME DO CAMPO     TIPO     TAMANHO
Modelo                          Text              15

Cor                                Text                8

Estado                           Text                2

Placa                              Text               7

Ano                                Integer

Valor                              Single

Após digitar os campos acima, feche essa janela, dando um click no botão “Close”.

Ao surgir novamente a janela “Table Structure” dar um click no botão “Build the Table” para gravar a tabela recém criada.

Após a gravação da tabela, vamos fazer a inclusão de registros. Para abrir a tabela VEICULOS e incluir os registros, dê um click com o botão direito do mouse no nome da tabela, que está na janela “Database Window”. Escolha o item “Open”.

Adicionar na tabela VEICULOS os registros abaixo, seguindo o seguinte procedimento:

a) Dar um click com o mouse no botão “Add”;

b) Digitar os conteúdos dos campos;

c) Clicar o botão “Update”;

d) Voltar ao passo “a”.

REGISTROS:

     MODELO          COR        ESTADO       PLACA      ANO       VALOR

1      FIAT UNO    BEGE             RJ             BUZ3107   2003        12000,00

2      GOL              VERDE           SP             BMJ1429    2001       13000,00

3      FIAT UNO     BRANCO       SP             BFB4538    2003       11500,00

4      CORSA          PRETO           SP             GWH8965  1999         9000,00

5      GOL               CINZA           MG            BFC3120    2002       15000,00

6      AUDI             AMARELO    SP              IGB6598    1996        15000,00

7      FIESTA          BRANCO       RJ              BMC2749  1996         6500,00

Após inserir todos os registros e pressionar pela última vez o botão “Update”, dar um click no botão “Close” e fechar todas as janelas, inclusive a do Visdata.

Obs.: O ACCESS é diferente de bancos de dados como o CLIPPER, DBASE, etc.

Para o ACCESS, um DATABASE pode conter várias “TABLES”.

Em nosso exemplo acima, o DATABASE “REVENDA” poderia conter as TABLES “VEICULOS”, “CLIENTES”, “CONTAS”, etc.  Se trabalhássemos com o dBASE, por exemplo, seriam DATABASES distintos.

EXERCÍCIO: Vamos fazer um programa em VB para manipular o banco de dados criado acima.

Fazer a tela de acordo com o layout a seguir:

Obs.: Nesta aplicação vamos utilizar dois formulários. Este é o primeiro formulário.

Obs.:
- Os objetos que conterão os campos do banco de dados, são Caixas de Texto (TextBox), pois iremos digitar valores neles (Por isso não podem ser Labels).

- O controle que nos vai permitir acessar o banco de dados é o “Data Control”, que 

aparece no rodapé do formulário, com suas setas para navegação pelo Banco de Dados.  As propriedades importantes que devemos configurar nesse controle, são as seguintes:

A configuração das seguintes propriedades do “Data Control” o ligam ao Banco de Dados: 

Obs. Note que não é preciso digitar os dados abaixo. Basta abrir os (...) na caixa de propriedades que o VB lista os valores existentes para os selecionarmos.

Configurar o Data Control: 

DATABASENAME=A:\REVENDA.MDB

RECORDSOURCE=VEICULOS

NAME=DATA1 

- Agora precisamos fazer a interligação das Caixas de Texto aos campos do Banco de Dados. Isso é feito configurando-se as seguintes propriedades das Caixas de Texto, que as ligam aos campos do Banco de Dados :

Obs. Note que não é preciso digitar os dados abaixo. Basta abrir os ( ...)  na caixa de propriedades que o VB lista os valores existentes para os selecionarmos.

Configurar cada uma das caixas de texto: 

DATASOURCE=DATA1

DATAFIELD=MODELO

DATASOURCE=DATA1

DATAFIELD=COR

DATASOURCE=DATA1

DATAFIELD=ANO

   . . . 

Após feita a configuração apontada acima, execute o programa e verifique que ele já accessa o banco de dados REVENDA e a tabela VEICULOS.  Você já pode visualizar os registros por meio do objeto “Data Control”.

Vamos melhorar o programa mais um pouco, digitando o seguinte código associado a este formulário:

                      Option Explicit

Private Sub Cmdeleta_Click ()

     data1.Recordset.Delete

     data1.Recordset.MoveNext

End Sub

Private Sub Cmdinsere_Click ()

     data1.Recordset.AddNew

     Txtmodelo.SetFocus

End Sub

Private Sub Cmdseleciona_Click ()

     form2.Show 1             ‘OBS. PODE SER 0=NÃO MODAL OU 1=MODAL

End Sub                           ‘VERIFIQUE A DIFERENÇA!...

Private Sub CmdNOk_Click ()

     data1.UpdateControls  ‘ATUALIZA APENAS A TELA DO MICRO

End Sub                            ‘OU SEJA, NÃO EFETUA AS ALTERAÇÕES

Private Sub CmdOk_Click ()

     data1.UpdateRecord  ‘ATUALIZA O BANCO DE DADOS

End Sub                          ‘OU SEJA, EFETUA AS ALTERAÇÕES NO BD.

Insira por meio do programa que você fez, mais alguns registros através do botão “Insere” e apague alguns registros por meio do botão “Delete”.

Insira os registros abaixo, que serão utilizados quando formos trabalhar com Crystal Report.

MODELO          COR        ESTADO       PLACA      ANO       VALOR

1      CORSA      BEGE             RJ             CAU3254     1998        8000,00

2      CORSA      VERDE          SP             BBJ2429      2000        12000,00

3      CORSA      BRANCO       BA            EBF9553      2002       15000,00

4      CORSA       PRETO          SP             GTH8865     1998        7500,00

Verifique se o programa está se comportando como esperado.

Note que esse programa está longe de ser uma aplicação profissional. Por exemplo, se você digitar valores inválidos nos campos, o programa vai aceitando tudo que se digita, até que ocorra algum erro irrecuperável e a execução é encerrada. Alguns desses problemas serão consertados daqui a pouco.

Vamos incrementar mais um pouco nosso projeto, incluindo o segundo formulário:

       Código associado a este formulário:

Option Explicit

Private Sub CmdCancela_Click ()

     FORM2.Hide    ‘ESCONDE O FORMULÁRIO, MAS CONTINUA

End Sub                 ‘ CARREGADO NA MEMÓRIA

Private Sub CmdOk_Click ()

     Form1.Data1.RecordSource = TxtExpSQL.Text

     Form1.Data1.Caption = TxtExpSQL.Text

     Form1.Data1.Refresh  ‘FORÇA A ATUALIZAÇÃO DO DATA CONTROL
                                        ‘COM OS PARÂMETROS DO SQL.

FORM2.Hide            

End Sub

OBS.:

O VB identifica e reconhece as instruções do SQL (Structured Query Language), que são compostas da seguinte forma:

SELECT ( nomes dos campos)

FROM ( nome do banco de dados)

WHERE ( condição ) 

Esperimente os seguintes exemplos, digitando-os  na janela “SELEÇÃO DE REGISTROS” do form2:

Select * from veiculos where ano > 80

Select * from veiculos where estado=“RJ”

Select * from veiculos Order By modelo

Select * from veiculos Order By cor Desc

Select * from veiculos where ano between 84 and 92

Select * from veiculos where modelo like “F*”

Select * from veiculos where modelo >= “F”

Obs.:

-Assim como accessamos o arquivo do ACCESS, podemos acessar arquivos de CLIPPER, FoxPro, Paradox, etc.

-Para isso, precisamos configurar adequadamente a propriedade “Connect” do “Data Control”.

-Se formos usar por exemplo um arquivo de dBase ou de Clipper, basta configurar a propriedade CONNECT=DBASE III. A partir daí o processo é o mesmo adotado para o caso do ACCESS.

EXERCÍCIOS PARA COMPLEMENTAÇÃO DA APLICAÇÃO “BANCO DE DADOS”:

1 Incluir na tela, um botão com Caption=“Fim” que, ao ser pressionado, encerra a execução do programa.O programa deve solicitar a confirmação de encerramento.

2 Incluir na tela principal, um botão com Caption “Seleciona Todos”, que, ao ser pressionado, seleciona todos os registros do banco de dados, em ordem ascendente de modelo. Tente as instruções:

Data1.RecordSource=”Select * From Veiculos”

Data1.Refresh

3 Incluir no seu projeto o controle “Microsoft Masked Edit Control”, e substituir a caixa de texto TxtValor por caixa de texto Masked Edit. Configure-a para receber o campo Valor e configure a propriedade Format para formato moeda.

4 Todos os dados dos veículos dever estar em maiúsculo. Isso pode ser feito usando a função Ucase e o evento LostFocus, ou o conjunto de funções : KeyAscii = ASC(UCASE(CHR(KeyAscii))), no evento KeyPress. 

5 Não deve permitir a digitação de caracteres alfa em campo numérico.

6 Limitar a digitação em cada campo, de acordo com a especificação do banco de dados (propriedade MaxLenght)

7 Validar o ano do veículo ( ANO >= 1900 and ANO <= YEAR (NOW) )

8 Tratar o erro no caso de deleção do último registro. A solução deve ser algo assim:

if Data1.RecordSet.EOF then Data1.RecordSet.MovePrevious

IMPRESSÃO DE RELATÓRIOS  UTILIZANDO O “CRYSTAL REPORTS”

Com a utilização do Crystal Reports obtém-se relatórios com detalhes muito bem elaborados e de fácil construção.

Os relatórios gerados pelo “Report Designer” ( Add-Ins/Report Designer) são gravados com a extensão “.RPT”, sendo que os dados que irão compor os detalhes do relatório devem vir de um arquivo de banco de dados.

Resumidamente, a emissão de um relatório usando o Crystal Reports consiste em:

-Elaborar o relatório usando o Report Designer;

-Incluir o Crystal Report Control na aplicação de onde se quer imprimir o relatório;

-Colocar na aplicação um controle para acionar o Crystal Report Control.

Vamos detalhar cada um desses passos por meio de um exemplo e em seguida por meio de um exercício.

Elaboração de um relatório usando o Report Designer:

- Escolha o item de menu Add-Ins/Report Designer;

- Abra o item File/New, e, na janela Create New Report dê um click em Standard;

- Na janela Create Report Expert, dê um click em Data File. Em seguida, selecione o caminho onde se encontra o arquivo de banco de dados. Em nosso caso, vamos utilizar o arquivo “Revenda”, que está no drive “a”. Após selecionar o arquivo, dê um click no botão Add. Se não houver outros arquivos a incluir, dê um click no botão Done.

- Selecione, a seguir, a guia “2: Fields”, para inserir os campos desejados no relatório. Inclua os campos Modelo e Cor. Em seguida, dê um click no botão “Preview”, para visualizar o relatório.

- A partir daí, por meio da guia “Design”, você pode elaborar melhor o seu relatório, incluindo campos especiais, tais como cabeçalhos, rodapé, data, número de página, outros campos do banco de dados, etc. 

- Digite um cabeçalho e um rodapé. Inclua no relatório, o campo “valor” ( Insert/Data Base Field), transportando esse campo para a área Details do relatório.

Para visualizar o relatório, selecione a guia “Preview” ou dê um click no ícone de visualização de pré impressão.

Salvar o arquivo com o nome A:RELVEIC, sendo que o Report Designer inclui a extensão “.RPT”.

A tela do projeto de nosso relatório, após incluirmos um cabeçalho, um rodapé e três campos no item Details, deve estar com a seguinte aparência:

Após visualizar o relatório, você já pode imprimi-lo diretamente da tela acima, escolhendo “File/Print/Printer” ou dando um click no ícone de impressão.

Obs.: Para que o relatório seja atualizado a cada vez que o programa é acionado, desligue a opção File / Save Report With Data.

CRYSTAL REPORT CONTROL 

Seu ícone na ToolBox:    [image: image22.png]



Podemos imprimir relatórios feitos pelo Report Designer em uma aplicação VB, por meio do Crystal Report Control.

Se esse controle não estiver na “ToolBox”, basta inclui-lo por meio do menu “Project/Components”, dando um click na Check Box do Crystal e no botão OK. 

Obs.: Esse controle não aparece em tempo de execução do programa.

Para exemplificarmos o uso desse controle, inicie um novo projeto e faça o layout do formulário de acordo com a janela abaixo: Esse formulário será, posteriormente incorporado ao seu projeto de banco de dados.

       E digite o seguinte código:

Private Sub Form_Load()

   CrystalReport1.ReportFileName = "a:relveic.rpt"  ‘Nome da estrutura do relatório.

   CrystalReport1.DataFiles(0) = "a:revenda.mdb"    ‘Nome do arquivo de Banco de Dados.

End Sub

Private Sub Cmdimprime_Click()

   CrystalReport1.Destination = 0 ‘0=janela, 1=impressora, 2=arquivo

   CrystalReport1.Action = 1     ‘ Dispara a impressão do relatório.

End Sub

Execute o programa e note que como atribuimos o valor 0 (zero)  à propriedade “Destination”, o relatório foi direcionado para a “Print Window” e não para a impressora.

A janela do relatório deverá estar mais ou menos como a seguinte:

Note que a “Print Window” ( janela acima ) possui vários ícones e opções que podemos utilizar imediatamente.

Algumas propriedades importantes do Crystal Report Control:

Action: Ao receber o valor Action=1, ocorre o disparo da impressão do relatório, para o destino escolhido ( propriedade Destination), que pode ser para a tela, impressora ou arquivo.

CopiesToPrinter: A quantidade de cópias desejada.

DataFiles: Especifica o caminho e os arquivos de banco de dados a serem utilizados no relatório.

Destination: Para onde irá o relatório: 0=Window; 1=Printer; 2=File.

PrintFileName: O nome do arquivo a ser gravado, se escolhermos Destination=2.

ReportFileName: O nome da estrutura do arquivo, com extensão .RPT.

WindowTitle: O Caption da barra de título da janela Print Window.

Exercício: Execute o “Report Designer”e  desenvolva um relatório parecido com o seguinte:

Observação: Para desenvolver este exercício, você pode criar um novo relatório ou incrementar o relatório anterior.

Campos especiais que podem compor este relatório:
-Cabeçalhos: Os campos tipo “Text” devem ser alinhados à esquerda e os campos numéricos devem ser alinhados à direita. Para isso, selecionar os campos e alinhá-los com os items Format/Field/Alignment;

-Para inserir os cabeçalhos em retângulos para destaque ou separação de campos, selecione o cabeçalho e use os itens Insert/Box;

-Logotipo: Faça o logotipo com o “Paint” e grave-o em disco. Em seguida, abra o item Insert/Picture, selecione o logotipo que você gravou e faça a inserção no cabeçalho do relatório.

-Inserção de Data : Insert/Special Field/Print_Date Field;

-Número de Página:Insert/Special Field/Page_Number Field;

-Totais e Subtotais: 

.Selecione o campo numérico desejado (Neste caso é VALOR) e abra os itens Insert/Subtotal;

.Escolher a ordem de classificação do relatório - neste caso é MODELO (Primeira quebra);

.Após o “OK”, seleciona-se novamente o campo numérico onde se deseja o subtotal (novamente o campo VALOR). Em seguida, escolhe-se o próximo campo de classificação - neste caso é ANO (segunda quebra);

Obs.: As quebras da maneira acima são uma dentro da outra, ou seja, o relatório é classificado  e sub-totalizado pelo campo MODELO. Porém, quando houver uma quebra de ANO para o mesmo MODELO, será feita TAMBÉM uma sub-totalização dos valores dos veículos daquele ANO, para aquele modelo.

.Inclusão de Total Geral : Seleciona-se o campo numérico que se quer totalizar (aqui também é selecionado o campo Valor ) e damos um click em  “Insert/Grand_Total”

.Inclusão dos textos ( Text Field) identificando os camposde totais e sub totais: “Sub Total de ano para este modelo:” 

O relatório a ser impresso deverá ter a seguinte aparência:

Em seguida, incremente a janela da aplicação que chama o “Crystal Report” para ficar como a janela a seguir e com o seguinte código (Obs. Option1 é uma matriz de três controles Option):

Código associado ao programa da página anterior:

Option Explicit

Dim tipojanela As Integer   ‘Variável global para seleção de onde será a impressão.

Private Sub Form_Load()

   CrystalReport1.ReportFileName = "a:relveic.rpt"  'Estrutura do Relatório

   CrystalReport1.DataFiles(0) = "a:revenda.mdb"     'Banco de Dados

   Option1(0) = True    ‘Deixa ligada a opção “Imprimir Para: Janela”

End Sub

Private Sub CmdImprime_Click()

On Error goto erro

   CrystalReport1.Destination = tipojanela '0=janela, 1=impressora, 2=arquivo

   If tipojanela = 2 Then                   ‘Faz a gravação em arquivo no disquete

      CrystalReport1.PrintFileName = "a:tstgrava.txt"   ‘Nome do arquivo a gravar

      CrystalReport1.PrintFileType = 2        ‘Verifique os outros valores possíveis

      Kill "a:tstgrava.txt"                         ‘Elimina o arquivo, se já existir...Cuidado: 

   End If                                                  ‘se o arquivo não existir, dará erro...

   CrystalReport1.Action = 1

   Exit Sub

Erro:  if Err = 53 then 

                  Resume next

          Else

                  MsgBox Error (Err)

          End if

End Sub

Private Sub Option1_Click(Index As Integer)

   tipojanela = Index

End Sub

Teste o programa que você fez para verificar se está tudo funcionando de acordo com o solicitado.

Por meio do Help On Line (F1), verifique os valores possíveis para a propriedade “PrintFileType”. Por meio do valor especificado nessa propriedade, você pode exportar os dados de seu relatório para ser usado em vários tipos de planilhas, vários tipos de arquivos texto, etc.

EXERCÍCIOS:

1- Salve este projeto com nome Imprime_relatorio;

2- Abra o projeto de banco de dados que foi feito anteriormente;

3- Incorpore ao projeto de banco de dados, o formulário “Impressão de Relatório”.

4- Inclua na tela principal do projeto de banco de dados, um botão com Caption “Relatório”, que ao ser pressionado, exibe, para ser executada de maneira modal, a tela “Impressão de Relatório”, que você acabou de incorporar ao projeto banco de dados. Teste o projeto todo para ver se tudo funciona.

5 – Modifique o formulário “Impressão de Relatório”  para que o usuário possa digitar a faixa de valores dos valores mínimo e máximo dos veículos a listar, bem como seu modelo.

Obs.:

a) Os campos de  valor mínimo e máximo só podem aceitar valores numéricos, exceto o ponto decimal.

b) Para restringir a lista de acordo com o item 4 acima, utilizar a propriedade SelectionFormula do Crystal Report Control, como no  exemplo:

Cryst1.SelectionFormula=”{veiculos.valor} >= 6000 and {veiculos.valor} <= 13000  and {veiculos.modelo} = ‘CORSA’” 

c) Para cancelar a seleção anterior, fazer = Cryst1.SelectionFormula=””   (carácter nulo).

PROGRAMA “CADASTRO DE COLEGAS”

O próximo exercício tem os seguintes objetivos:

- O aprendizado de uso do controle “Combo Box;

- O uso de um módulo em VB; 

- Criação e uso de subrotina

- A definição de um novo tipo de dado (comando Type);

- A impressão de dados através do método “print” sobre o objeto “printer”.

O programa fará o seguinte:

- Permitirá que você vá incluindo dados cadastrais : nome, endereço e  telefone em uma tabela na memória;

- Quando for dado um click para abrir a lista do quadro combo, os nomes que foram incluídos serão listados em ordem alfabética;

- Quando você selecionar um determinado nome na lista do quadro combo, ao dar um click nesse nome, os dados:  nome, endereço e telefone surgirão nas caixas de texto correspondentes.

- Quando você pressionar o botão “IMPRIMIR”, os conteúdos das caixas de texto serão enviados à impressora.

             Fazer a tela de acordo com o layout abaixo:

Obs.: O controle Combo Box deverá ter as propriedades Style=Dropdown List e Sorted=True, para que tenha a forma igual à da janela acima e para que apresente os dados em ordem alfabética. 

Código associado ao formulário anterior:                  

Option Explicit

Private Sub CMDIMPRIME_Click ()

                               ‘Note que na impressão com Printer.Print, podemos usar as funções

                               ‘ de espacejamento Tab() e Spc(), separadas por ponto e vírgula.

                               ‘Note também que podemos continuar uma instrução longa na linha

                               ‘seguinte, usando o símbolo de continuação “_”.

            ‘   PRINTER.NEWPAGE        ‘ Esta instrução faz saltar uma página   

             Printer.Print Spc(3); "Nome"; Tab(15); "Endereço"; Tab(35); _

                                              "Telefone"; Tab(50); "..... dados a seguir.."

             Printer.Print txtnome.Text; Tab(15); txtender.Text; _

                                               Tab(35); txtfone.Text

             PRINTER.EndDoc                         ' Esta instrução dispara a impressão

End Sub

Private Sub CMDINSERIR_Click ()

     COLEGAS(INDICE).NOME = TXTNOME.Text   ‘ INDICE é uma variável global

     COLEGAS(INDICE).ENDER = TXTENDER.Text

     COLEGAS(INDICE).FONE = TXTFONE.Text

     COMBO1.AddItem COLEGAS(INDICE).NOME  ' Incrementa a lista do “Combo”

     INDICE = INDICE + 1

     TXTNOME.Text = ""      ‘Limpa os campos, colocando o carácter nulo

     TXTENDER.Text = ""

     TXTFONE.Text = ""

     TXTNOME.SetFocus      ‘Posiciona o cursor neste campo

End Sub

Private Sub Combo1_Click ()

       PROCURA        ‘Chamada da Procedure Sub “PROCURA”

End Sub

A Sub PROCURA, que vem a seguir, deve ser codificada na área “declarations” do “GENERAL”, pois ela vai ser chamada por uma outra procedure, devendo pois, ser 

GLOBAL ao formulário.  Por esse motivo, ela começa com a palavra reservada “Public”.  As funções e procedures que não forem especificadas como “Private”, são “Public” por default.

 Public Sub PROCURA () ‘Esta procedure pesquisa a matriz COLEGAS, 

              ‘visando localizar uma string igual à que está no “text”do quadro combo.

   Dim I As Integer

   For I = 0 To INDICE - 1   ' Indice é Variável Global.

         If COMBO1.Text = COLEGAS(I).NOME Then   ‘quando localizar, passa os

              TXTNOME.Text = COLEGAS(I).NOME        ‘campos:  nome, ender, fone

              TXTENDER.Text = COLEGAS(I).ENDER      ‘para as caixas de texto.

              TXTFONE.Text = COLEGAS(I).FONE

         End If

   Next I

End Sub

Inclusão de um módulo ao projeto... 

Lembre-se que um módulo é global às procedures de todo o projeto. Para incluir um módulo ao projeto, abra o item “Project / Add Module”. Em seguida, digite o código abaixo:

Option Explicit

Type CADASTRO                          ‘Definição de um novo tipo de dado.

      NOME As String * 20

      ENDER As String * 20

      FONE As String * 11

End Type

Public COLEGAS(0 To 100)  As CADASTRO   ‘A matriz COLEGAS é do tipo

                                                                              ‘ CADASTRO, definido acima.

Public INDICE As Integer

Poderíamos incluir aqui funções que possam ser chamadas de qualquer local do projeto.
Execute o programa, incluindo alguns nomes para teste. Lembre-se que como o arquivo (matriz COLEGAS) está na memória, se você encerrar o programa, todos os dados que foram incluídos serão perdidos. O manuseio de arquivos em disco, será visto posteriormente.

Exercícios:

1- Modificar a sub PROCURA que você codificou no “general” do formulário FORM1 e transportá-la para dentro do MODULE1. Não esqueça de apagá-la do FORM1.

ATENÇÃO!  Como as funções codificadas dentro dos módulos são globais ao projeto, quando se referencia algum objeto nessas funções, há necessidade de referenciarmos o nome do formulário a que esse objeto pertence. 

Por exemplo, dentro do MODULE1,  a referência aos objetos do FORM1, serão: 

If FORM1.COMBO1.Text = COLEGAS(I).NOME Then     ‘quando localizar, passa os

              FORM1.TXTNOME.Text = COLEGAS(I).NOME ‘campos: nome, ender, fone

              FORM1.TXTENDER.Text = COLEGAS(I).ENDER   ‘para as caixas de texto.

              FORM1.TXTFONE.Text = COLEGAS(I).FONE

End If

2- Note que o programa insere nomes em branco, sendo que isso não é desejável. Para que isso não aconteça, verifique se o nome está em branco. Se estiver, o programa deve exibir uma mensagem de advertência (Use  a Msgbox) para indicar o erro e não fazer a inserção. 

3- Modificar a sub PROCURA para abandonar a malha (sair do processo iterativo), quando encontrar o nome procurado 

4- Modificar o programa para que não permita a inclusão de nomes iguais.

Obs. Cuidado quando fizer a comparação do TxtNome.Text com Colegas(i).Nome, pois seus tamanhos são incompatíveis. Nesse caso, talvez seja mais conveniente mover o conteúdo de TxtNome.Text para uma variável declarada como String*20, para comparar com Colegas(i).Nome,. que têm essas mesmas características.
IDENTIFICAÇÃO DE CORES

Nosso próximo trabalho nos habilitará a identificar regiões da tela por mais irregulares que elas sejam.

Com certeza você já  viu um programa que exibe um mapa de um país por exemplo, com contornos bastante irregulares sendo que quando se dá um click sobre uma cidade ou um estado, o programa identifica a região apontada e informa detalhes sobre aquele local.

Fazer a localização pelas coordenadas do local apontado seria muito difícil pelo fato da região ser geralmente irregular, não sendo possível a obtenção de um algoritmo para isso.

Um método bastante utilizado é a identificação da cor utilizada naquela região. Note que na maioria desses aplicativos, as regiões são pintadas de cores diferentes. A identificação é feita pela cor do local apontado sendo em alguns casos combinando-se a cor com o quadrante em que as figuras estão divididas.

Em nosso próximo exercício, iremos identificar alguma figuras, exibindo uma caixa de mensagem (Message Box) informando em qual figura foi dado o click com o mouse.

Para fazer esse trabalho, vamos desenhar no Paintbrush ou no Paint, algumas figuras como no desenho abaixo e inserir o arquivo com o desenho, na propriedade Picture do Formulário.

A tela de seu programa deverá estar parecida com a abaixo:

Em seguida, para uma primeira identificação das cores vamos fazer um programa como o abaixo:

Obs.: Verificar com o help on line (F1) o significado dos argumentos da sub que trata o evento “MouseDown”. Rapidamente sem grandes detalhes, vamos descrevê-los:

Button: Retorna um valor numérico indicando qual botão do mouse foi pressionado. Se o valor for 1, foi pressionado o botão esquerdo, 2 foi o botão do centro e 4 foi o botão direito.

Shift: Este parâmetro conterá um valor numérico indicando se o botão foi pressionado em  conjunto com uma das teclas Shift (conterá o valor 1), Ctrl (conterá o valor 2), Alt (conterá o valor 4).

X e Y conterão as coordenadas do ponto em que estava o apontador do mouse quando houve o click, sendo X o ponto das abcissas e Y o ponto das ordenadas.

ATENÇÃO: O PAINTBRUSH NÃO É MUITO BOM PARA ESSE TIPO DE TRABALHO POIS A MAIORIA DAS CORES DISPONÍVEIS SÃO COMBINAÇÕES DAS CORES BÁSICAS (RGB - Red Green Blue), NA FORMA DE PEQUENOS RETÂNGULOS DE CORES DIFERENTES, QUE QUANDO VISTOS EM FORMA MACROSCÓPICA, FORMAM AS MAIS VARIADAS CORES, TAIS COMO O VERDE LIMÃO, O ALARANJADO, ETC. CADA UM DESSES PEQUENOS RETÂNGULOS É CHAMADO DE “PIXEL”. 

ISSO PODE SER VERIFICADO UTILIZANDO-SE O ITEM “VISUALIZAR - MAIS ZOOM” NO MENU DO PAINTBRUSH. SE A FIGURA FOR COMPOSTA DE “PEQUENOS QUADRADOS” (PIXÉIS) DE CORES DIFERENTES, DEVE-SE ESCOLHER UMA NOVA COR, OU CONSIDERAR-SE NO PROGRAMA PARA EFEITO DE IDENTIFICAÇÃO DA FIGURA, TODAS AS CORES BÁSICAS QUE  FAZEM PARTE DA COR MACROSCÓPICA..

O MOTIVO DISSO É QUE QUANDO OCORRE O CLICK DO MOUSE NA ÁREA DELIMITADA PELA MESMA COR (OU QUE PELO MENOS APARENTA A MESMA COR NO  FORMATO MACRO), SERÃO RETORNADOS VALORES DIFERENTES PELA CHAMADA DA FUNÇÃO Point(X,Y), POIS MICROSCOPICAMENTE O APONTADOR DO MOUSE ESTARÁ POSICIONADO ALTERNATIVAMENTE EM PÍXEIS DE CORES DIFERENTES.  

Código inicial em VB:

Option Explicit

Private Sub Form_MouseDown (Button As Integer, Shift As Integer, X As Single, Y As _

                                                                                                                         Single)

      Dim COR As Long

      COR = Point(X, Y) ‘ RETORNA A COR  RGB DO PONTO DE CORDENADAS (X, Y) 

                                     ‘ ONDE FOI DADO O  CLICK DO MOUSE.

      MsgBox "COR = " + Str(COR), 48, "A T E N Ç Ã O"  ‘EXIBE O VALOR NUMÉRICO

                                                                                            ‘DA COR - INTEIRO LONGO.

End Sub

****  EXECUTE O PROGRAMA E ANOTE EM UM PAPEL OS VALORES RETORNADOS DA MSGBOX, PARA INCLUSÃO DESSES VALORES  NA FUNÇÃO CLICOU() QUE VOCÊ FARÁ EM SEGUIDA.

Obs.: Os valores numéricos que aparecem no programa abaixo não serão necessariamente aqueles de seu programa.  Verifique os valores que você anotou para as figuras de sua aplicação e coloque-os adequadamente em seu programa.

A função Point(X,Y) retorna um inteiro longo, de 4 bytes, tendo como conteúdo o valor RGB da cor no ponto onde ocorreu o click do mouse.

Esse inteiro longo tem a seguinte forma:

              00      R       G      B   

                byte1   byte 2    byte3    byte4

Digitar a Função CLICOU () abaixo. Note que você pode digitá-la em qualquer local da janela de código, que o VB a transfere para a área “General”.

Function CLICOU$ (COR As Long)

  If COR = 255 Then  CLICOU$ = "VOCÊ CLICOU NO CÍRCULO"

  If COR = 6280 Then  CLICOU$ = "VOCÊ CLICOU NO TRIÂNGULO"

  If COR = 16711680 Then  CLICOU$ = "VOCÊ CLICOU NO QUADRADO"

  If COR = 16777215 Then CLICOU$ = "VOCÊ CLICOU FORA DAS FIGURAS"

     '  ou poderia ser:

     '  select case cor

     '            Case 255

     '                   CLICOU$ = "você clicou  no  círculo"

     '                   Shell ("sndrec32.exe  /play  /close  c:\windows\msremind.wav")

     '            Case 6280

     '                    CLICOU$ = "você clicou no triângulo"

     '                    Shell ("sndrec32.exe  /play  /close  c:\windows\ding.wav")

     '           . . .

     '  end select

End Function

Obs.: A função Shell do exemplo anterior permite que se execute qualquer aplicativo externo, a partir do programa em VB. No exemplo, estamos executando o programa SNDREC32.EXE, que executa arquivos sonoros ( com extensão .WAV).

A execução da instrução Shell ("sndrec32.exe  /play /close  c:\windows\msremind.wav"), provoca a execução do programa SndRec32, e, logo após a execução da música contida no arquivo MsRemind.Wav, o programa SndRec32 é encerrado.

Em seguida, inclua um LABEL de nome LBLFIGURA na parte inferior da tela de sua aplicação para que seja exibida a mensagem da figura em que foi dado o click do mouse e altere a Private Sub Form_MouseDown para ficar como abaixo:  

Option Explicit

Private Sub Form_MouseDown (Button As Integer, Shift As Integer, X As Single, Y As _ 

                                                                                                                      Single)

     Dim COR As Long

     COR = Point(X, Y) ‘ RETORNA A COR  RGB DO PONTO DE CORDENADAS (X, Y) 

                                     ‘ ONDE FOI DADO O  CLICK DO MOUSE.

     ‘ MsgBox "COR = " + Str(COR), 48, "ATENÇÃO"  ‘EXIBE O VALOR NUMÉRICO

                                                                                            ‘DA COR - INTEIRO LONGO.
      LBLFIGURA.Caption = CLICOU$(COR)  ‘ EXIBE A MENSAGEM 

End Sub

Assim como você exibe uma simples mensagem sobre a figura, poderia emitir uma série de informações. Por exemplo se a figura fosse o mapa do Brasil com seus estados, poderia exibir a bandeira do Estado, informar sua população, pontos turísticos, hinos, músicas típicas etc.

Exercício: Fazer com que ao ocorrer o click do mouse, a cor de fundo do label, fique igual à cor onde ocorreu o click.

MANUSEIO DE ARQUIVOS

O VB suporta 3 tipos de arquivos: SEQUENCIAL, ALEATÓRIO e BINÁRIO.

Permite a abertura de arquivos de 3 maneiras: Entrada (Input), Saída( Output) e Acréscimo (Append).

Nosso próximo exercício vai explorar apenas arquivos sequenciais, onde iremos desenvolver um formulário que você poderá utilizar em seus próximos programas que fazem leitura de arquivos.

Esse formulário é composto pelos objetos DriveListBox, DirListBox e FileListBox, que quando interligados permitem a seleção de Drives, Diretórios e Arquivos.

O exercício a seguir é composto de 3 formulários, sendo:

Primeiro formulário (Form1):

Projetar  o formulário acima com uma caixa de textos de nome “Txtexto”, especificando as propriedades “MultLine=True” para que quando se pressione a tecla Enter, o cursor passe para a linha seguinte. Atribua à propriedade “ScrollBars=Both” para que aparecam as barras de rolagem vertical e horizontal.

O finalidade desse formulário é receber na caixa de texto, o conteúdo de um arquivo sequencial que será lido quando se dá um click no botão “Ler”, ou enviar o texto para um arquivo, quando ocorrer o click no botão “Gravar”.

Código em VB associado a esse formulário:

Option Explicit

Private Sub CMDgravar_Click ()

    form2.show 1

End Sub

Private Sub CMDler_Click ()

    form3.show 1

End Sub

 Note que o código acima limita-se a chamar o “form2” no caso de gravação de arquivo ou a chamar o “form3” no caso de leitura. 

Como exercício, inclua no formulário anterior, o botão “SAIR”.

           ( A instrução que encerra a aplicação VB é apenas “end”).

Portanto, vejamos o segundo formulário ( Form2) que providencia a gravação do arquivo.

A janela acima surge quando se dá um click no botão “Gravar” do primeiro formulário. Na caixa de texto “TxtArqnome”, digitamos o caminho e o nome do arquivo que queremos gravar, como por exemplo “A:\MEUDIR\MEUARQ.TXT”  e em seguida damos um click no botão “OK”.

Vejamos agora a programação associada ao formulário acima: 

Option Explicit

Private Sub CmdCancela_Click ()

    Form2.Hide

End Sub

Se dermos um click no botão “CmdCancela”, simplesmente a janela acima ficará escondida, sendo novamente ativado o primeiro formulário.

Obs.: O método HIDE, apesar de esconder o formulário, não o desativa nem descarrega da memória. O formulário está apenas escondido sendo que seus objetos podem ser referenciados.

Por exemplo, no caso acima após a execução da instrução “Form2.Hide”, poderíamos acessar do primeiro formulário, o conteúdo da caixa de texto 

TxtArqNome. Poderíamos ter lá no Form1 a instrução “Print Form2.TxtArqNome.Text”. Como o Form2 ainda está na memória (apesar de escondido), o conteúdo da caixa de texto do Form2 será impresso na janela do Form1.

Continuando, vamos analisar algumas novidades no código em VB abaixo, que será executado quando ocorrer o click no botão “CmdOk”.

A primeira delas é a instrução “On Error Goto ...”. Essa instrução é bastante utilizada em VB sendo muito útil em Subs e Functions onde há possibilidade de ocorrência de erros tais como  erros de entrada de dados, gravação e leitura de arquivos, erros decorrentes de cálculos matemáticos tais como estouro de campos, divisão por zero, etc.

Para verificar o formato da instrução, selecione “On Error” e pressione F1 para ativar o Help on line. 

O restante do programa faz basicamente o manuseio de um arquivo, que é comum a qualquer tipo de arquivo em qualquer linguagem, que consiste na abertura do arquivo, a gravação dos dados no arquivo e o fechamento, que são feitos em VB, pelas instruções Open, Print e Close.

Private Sub CmdOk_Click ()

On Error GoTo Erro

    Open TxtArqNome.Text For Output As #1

    Print #1, Form1.Txtexto.Text

    Close #1

    Form2.Hide

    Exit Sub

Erro:

    MsgBox "ERRO NO ARQUIVO...", 48, "A R Q U I V O S"

    Resume Fim

Fim:

End Sub

O terceiro formulário é um pouco mais complexo pois envolve três novos objetos, que quando interligados nos permitirão selecionar drives, diretórios e finalmente o arquivo que queremos ler:

                    Ícones da caixa de ferramentas:        [image: image32.png]


   

O primeiro deles é o DriveListBox que nos permite selecionar o drive onde se encontra o arquivo, o segundo é o DirListBox para seleção do diretório e finalmente FileListBox onde selecionamos o arquivo desejado.

Esquematicamente, podemos visualizar o processo da seguinte maneira:

                                         EVENTO CHANGE                   EVENTO CHANGE
          UNIDADE                                           DIRETÓRIO                                 ARQUIVO

                                                                                                                          FILE1.PATH

      DRIVE1.DRIVE                                DIR1.PATH                               FILE1.FILENAME 

          DriveListBox                                     DirListBox                                  FileListBox

Ao ocorrer o click do mouse alterando a unidade de disco ( DriveListBox) , ocorrerá o evento Change que acarretará alterações também na caixa de diretórios (DirListBox), ocasionando alterações na caixa de arquivos (FileListBox). Isso é feito pela seguinte “Sub”:

Private Sub Drive1_Change ()

On Error GoTo erro

   Dir1.Path = Drive1.Drive

   Exit Sub

erro:

   MsgBox "Dispositivo inacessível"    ‘EM CASO DE ERRO, VOLTA AO DRIVE

   Drive1.Drive = Dir1.Path           ‘QUE FOI SELECIONADO ANTERIORMENTE.

End Sub

Se o click do mouse for na caixa de diretórios, acarretará alterações apenas nela própria e na caixa de arquivos.  Isso é feito pela seguinte “Sub”:

Private Sub Dir1_Change ()

    File1.Path = dir1.Path

End Sub

Se o click do mouse for na caixa de arquivos, as modificações ocorrerão apenas nela.

Execute o programa após a codificação acima e verifique que a seleção de drives e diretórios já funciona pois já foi feita a amarração entre os objetos que os controlam.

Segundo o exposto acima, o programa completo para este formulário, ficará assim:

Option Explicit

Private Sub CmdCancela_Click ()

    Form3.Hide                  ‘ESCONDE O FORM3, VOLTA O ANTERIOR.

End Sub

Obs.: Quando houver o click no botão “Ok”, será feita a concatenação  do nome do arquivo ao “Path”, completando o caminho da localização do arquivo.

Private Sub CmdOk_Click ()

Dim nomearq As String

On Error GoTo Erro

    If (Right$(dir1.Path, 1) = "\") Then

        nomearq = dir1.Path + File1.FileName   ‘JÁ CONTEM A “\”

    Else

        nomearq = dir1.Path + "\" + File1.FileName ‘NÃO CONTÉM A “\”.

    End If                                                                 ‘NESSE CASO, É INCLUÍDA 

    Open nomearq For Input As #1

    form1.Txtexto.Text = Input$(LOF(1), #1) ‘PEGA O CONTEÚDO DO ARQUIVO.

                                                                      ‘LOF(1) É O TAMANHO DO ARQ. 

    Close #1

    Form3.Hide

    Exit Sub

Erro:

    MsgBox "Erro no arquivo...", 48, "A R Q U I V O"

    Close #1                  ‘  fecha o arquivo, para poder abrir depois, novamente

Resume Encerra                   

Encerra:

End Sub

Private Sub Drive1_Change ()

On Error GoTo erro

   Dir1.Path = Drive1.Drive

   Exit Sub

erro:

   MsgBox "Dispositivo inacessível"    ‘EM CASO DE ERRO, VOLTA AO DRIVE

   Drive1.Drive = Dir1.Path           ‘QUE FOI SELECIONADO ANTERIORMENTE.

End Sub

Private Sub Dir1_Change ()

    File1.Path = dir1.Path

End Sub

Fazer vários testes com o programa para verificar se está de seu agrado.

Exercício:

Incluir um quadro Combo no  “Form3” onde você possa selecionar arquivos tais como “*.*”, “*.doc” ou “*.txt”.

Isso pode ser feito com a inclusão de um quadro combo “Combo Box Control” em nosso formulário (Form3) e o seguinte código:

Private Sub Combo1_Click ()

    file1.Pattern = combo1.Text     ‘SELECIONA O TIPO DE ARQUIVO

End Sub

Private Sub Form_Load ()

    combo1.AddItem "*.*"    ‘INCLUI UM ITEM NA LISTA DO COMBO

    combo1.AddItem "*.doc"      ‘Pode-se também usar a propriedade “List” e

    combo1.AddItem "*.txt"        ‘fazer a inclusão de itens em tempo de projeto.

    combo1.ListIndex = 0 ‘Posiciona o texto na posição 0 (no texto *.*)

End Sub

Obs. A propriedade “ListIndex” de ComboBox é muito utilizada no relacionamento entre uma tabela, por exemplo, e os itens da lista do Combo. Quando ocorre o click do mouse em um item da lista do combo, na propriedade ListIndex, retorna o número sequencial do item onde ocorreu o Click. O valor da propriedade ListIndex , inicia com o número 0 (zero) e vai até n-1 elementos da lista..

A forma final da tela para o acesso de arquivos, deverá ficar de acordo com o abaixo:

Exercícios:

1 - Coloque uma “pista” (hint)  no botão “Ok” explicando o que faz esse botão. Coloque outra “pista” no botão “Cancela”. A “pista”deve surgir quando o apontador do mouse estiver sobre o objeto que o programador quer exibir a “pista” (no caso, os botões Ok e Cancela). 

Isso é feito utilizando-se a propriedade ToolTipText. 

2 - Nas aplicações Windows que manipulam arquivos, a abertura de arquivos geralmente e executada ao ocorrer o click duplo (evento DoubleClick) sobre o nome do arquivo, executando o mesmo procedimento de selecionar o arquivo e dar um click no botão Ok.

Programe o evento DoubleClick para que isso ocorra em nossa aplicação.

Obs.: Isso pode ser feito de duas maneiras:

1-Copiando o programa da procedure CmdOk_Click e colando-o no evento DblClick do controle File1;

2-Executando diretamente, no evento DblClick do controle File1, a procedure CmdOk_Click.

Isso é feito escrevendo-se diretamente o nome da procedure, que a mesma será executada:

Private Sub File1_DblClick()

  CmdOk_Click

End Sub

Gravar este formulário com um nome diferente, como por exemplo LEARQS.FRM. 

Sempre que você precisar fazer leitura de arquivo, o que é muito comum em aplicações profissionais, você poderá chamá-lo e incluí-lo já pronto em seu projeto.

Este formulário deverá ser novamente utilizado por nós posteriormente em outro exercício.

Observações importantes :

A propriedade “name” do formulário é muito importante sendo sempre interessante que o “name” do formulário tenha relação com a função que o mesmo desempenha.

Quando se grava o formulário em disco ( File/Save Formn AS), o nome que se atribui ao arquivo é o seu NOME EXTERNO, sendo que o nome interno ( o “name” do formulário) não é necessariamente igual.

Se você for reutilizar formulários já feitos, o que é muito comum em programação, certamente terá problemas se não der nomes (names) sugestivos aos seus formulários. Manter os nomes Form1, Form2, etc., atribuídos pelo VB, só lhe trará confusões e trabalhos extras.

Se você tentar incluir em seu projeto, um formulário feito anteriormente, e, o nome interno (name) dos formulários forem iguais ( por exemplo Form1), o VB dará uma mensagem de erro (não muito amigável), e não aceitará a inclusão.

Portanto, verifique qual o “name” do formulário que você acabou de fazer e lhe dê um nome adequado, como por exemplo LEARQS.   Não há nenhum problema em termos um nome interno igual ao nome externo.

O CONTROLE “GRID”
Em nosso próximo exercício utilizaremos o controle “Grid”, que não faz parte do conjunto padrão de ferramentas do VB 5.0. O controle Grid é bastante rudimentar quando comparado com outras ferramentas similares encontradas no mercado, porém para mostrarmos a utilidade desse tipo de ferramenta, será suficiente.

O controle Grid é utilizado quando se deseja separar informações tais como dados numéricos, figuras, textos, etc., em forma de tabelas. Seu tratamento e uso é muito parecido com o de uma planilha eletrônica.

Para exemplificar, vamos fazer uma aplicação em que construiremos uma planilha eletrônica simplificada.

Caso o controle Grid não esteja em sua Toolbox, inclua-o da seguinte maneira: selecione os itens do menu “Project/Components/Microsoft Grid Control”, ou “Project/Components/Microsoft Flex Grid Control”.

Após pressionar o botão Ok, verifique que o ícone do Grid Control foi incluído em sua ToolBox.

Para demostrarmos o uso desse controle, vamos elaborar uma aplicação como a descrita abaixo, incluindo nela o controle “Grid”, com nome (name) de “GrdTab”.

Para obter 5 linhas e 6 colunas, atribuir às propriedades: ROWS=5 e COLS=6.

Para que a figura (Picture1) não apareça em tempo de execução, fazer VISIBLE=FALSE.

Código associado à aplicação acima:
Inicialmente, no evento “Form Load” iremos inicializar com cabeçalhos de linhas, colunas e 

iremos chamar 3 procedimentos que aumentam linhas, aumentam colunas e inicializam células:

Private Sub Form_Load ()

     grdtab.Row = 0  ‘ OBS.:AS LINHAS E COLUNAS COMEÇAM EM 0.

     grdtab.Col = 1

     grdtab.Text = "Água"

     grdtab.Col = 2

     grdtab.Text = "Luz"

     grdtab.Col = 3

     grdtab.Text = "Comida"

     grdtab.Col = 4

     grdtab.Text = "Combustível"

     grdtab.Col = 5

     grdtab.Text = "Total"

     grdtab.Row = 1

     grdtab.Col = 0

     grdtab.Text = "Janeiro"

     grdtab.Row = 2

     grdtab.Text = "Fevereiro"

     grdtab.Row = 3

     grdtab.Text = "Março"

     grdtab.Row = 4

     grdtab.Text = "Total"

     aumentacol                      ‘CHAMADA DE PROCEDURE 

     aumentalin                       ‘CHAMADA DE PROCEDURE

     preenche                          ‘CHAMADA DE PROCEDURE

End Sub

Em seguida, vamos fazer as procedures AUMENTACOL, AUMENTALIN E PREENCHE:

Public Sub aumentacol ()

      Dim i As Integer

      For i = 0 To 5

           grdtab.ColWidth(i) = 1300  ‘VALOR EM TWIPS

      Next i

End Sub

Public Sub aumentalin ()

      Dim i As Integer

      For i = 0 To 4

           grdtab.RowHeight(i) = 400 ‘VALOR EM TWIPS

      Next i

End Sub

Public Sub preenche ()

     Dim i, j As Integer

     For i = 1 To 4

          grdtab.Row = i

          For j = 1 To 5

                grdtab.Col = j

                grdtab.Text = "NADA"   ‘PREENCHE TODA A PLANILHA

          Next j

     Next i

End Sub


Quando for dado um click no botão “Limpa”, deverá limpar todas a células. Para isso, será executado o código abaixo: 

Private Sub cmdlimpa_Click ()

     limpa

End Sub

Public Sub limpa ()

     Dim i, As integer, j As Integer

     For i = 1 To 4

          grdtab.Row = i

          For j = 1 To 5

                grdtab.Col = j

                grdtab.Text = ""   ‘PREENCHE COM O CARÁCTER NULO
          Next j

     Next i

End Sub

Ao executar o programa, note que não adianta pressionarmos teclas que os valores não serão inseridos nas células. Para digitarmos valores nas células, precisamos tratar os eventos que ocorrem nas células. Verifique o código abaixo:

Private Sub grdtab_KeyPress (keyascii As Integer)

    Dim tecla As String

    Dim soma As Single

    Dim i as integer, j as integer, oldlin as integer, oldcol  As Integer

                                          ‘Só ACEITA VALORES NUMÉRICOS  

    tecla = Chr$(keyascii)

    If (tecla < "0" Or tecla > "9") And tecla <> "." Then

        Beep

        keyascii = 0

    End If

                                           ‘ACRESCENTA O NÚMERO DIGITADO NO

                                           ‘ FINAL DO STRING

    grdtab.Text = grdtab.Text + Chr$(keyascii)

    oldlin = grdtab.Row     ‘MEMORIZA A LINHA PARA VOLTAR DEPOIS

    oldcol = grdtab.Col       ‘MEMORIZA A COLUNA PARA VOLTAR DEPOIS

    soma = 0

    For i = 1 To 3                   ‘EFETUA A SOMA DA COLUNA

        grdtab.Row = i

        soma = soma + Val(grdtab.Text)

    Next i

    grdtab.Row = 4

    grdtab.Text = Str$(soma)    ‘COLOCA O VALOR DA SOMA NA CÉLULA

    grdtab.Row = oldlin ‘VOLTA PARA A LINHA QUE HAVIA MEMORIZADO

End Sub

 Pode-se inserir figuras nas células, como por exemplo: 

Private Sub CmdInserePict_Click ()

 grdtab.Picture = pic1.Picture

End Sub

Obs.: O VB, por meio do comando “LOADPICTURE” permite que se insira em vários objetos,  figuras que estejam armazenadas um meio magnético qualquer. Por exemplo, poderíamos fazer a Private Sub CmdInserePict_Click () da seguinte maneira:

Private Sub CmdInserePict_Click ()

grdtab.Picture=LOADPICTURE(“C:\arquiv~1\devstu~1\vb\graphics\icons\elements\rain.ico”)

End Sub

Usando o artifício acima, você poderia incluir em sua aplicação um Quadro Combo onde o usuário poderia selecionar qual figura quer incluir, por meio de seu nome.

Se quiser remover figuras, usar o LOADPICTURE sem argumentos:

         GrdTab.Picture=LoadPicture()

Exercícios: 

Incluir um quadro combo, inicializado com 4 nomes de figuras, como por exemplo: FOGO, AGUA, CHUVA, NEVE. O operador deverá escolher E dar um click em um dos nomes da lista, para que a figura correspondente seja incluída na célula que está selecionada na planilha (vide  exemplo anterior de uso do quadro combo).

Inserir os nomes na lista do quadro Combo em tempo de projeto, usando a propriedade “List”.  Neste exercício, seu programa deve ficar mais ou menos como o programa abaixo:

Private Sub Combo1_Click()

  Select Case Combo1.ListIndex

   Case 0

     grdtab.Picture = LoadPicture("C:\arquiv~1\devstu~1\vb\graphics\icons\elements\fire.ico")

     grdtab.Text = "Fogo"

   Case 1

     grdtab.Picture=LoadPicture("C:\arquiv~1\devstu~1\vb\graphics\icons\elements\water.ico")

     grdtab.Text = "Água"

    Case 2

     grdtab.Picture = LoadPicture("C:\arquiv~1\devstu~1\vb\graphics\icons\elements\rain.ico ")

     grdtab.Text = "Chuva"

    Case 3

     grdtab.Picture=LoadPicture("C:\arquiv~1\devstu~1\vb\graphics\icons\elements\snow.ico ")

     grdtab.Text = "Neve"

  End Select

End Sub
O controle Grid permite manusear células selecionadas, como no exemplo de impressão abaixo:

Private Sub cmdprint_Click ()

     Print grdtab.SelStartRow

     Print grdtab.SelEndRow

     Print grdtab.SelStartCol

     Print grdtab.SelEndCol

End Sub

Obs. Quando se usa o método Print como no exemplo acima, a impressão é feita no próprio Formulário. Se quisermos imprimir na impressora, devemos ativar o Objeto Printer.

Exercício: Incluir código em VB para totalizar todas as linhas e colunas, à medida que se vai digitando os números nas células. 

GERAÇÃO DE NÚMEROS ALEATÓRIOS.

O VB possui a instrução “Randomize” que em conjunto com a função RND gera números aleatórios. Verifique o programa a seguir:

Private Sub Form_Click()

     Dim I As Integer

     Dim aleatorio As Single

     Randomize    'Inicia uma nova sequência (seed)

     aleatorio = Rnd  'gera um valor aleatório entre 0 e 1

     I = 6 * aleatorio 'faz o número ficar entre 0 e 6, inteiro

     Print I, aleatorio 'imprime os valores

 End Sub

Execute o programa e note que a cada click sobre o formulário, é impresso um valor aleatório, no intervalo de    0   a    6.

Exercício: Utilizando as funções LoadPicture, Randomize e RND, fazer um programa que ao ser executado, vai trazendo, de 1 em 1 segundo, figuras aleatórias (de um total de 6 figuras) em um controle “image”.

Dica: Utilizar uma variável randomizada “I” e a estrutura Select Case.

O programa poderá ficar mais ou menos como abaixo:

Private Sub Timer1_Timer()

  Dim I As Integer

  Randomize    'Inicia uma nova sequência (seed)

  I = 6 * Rnd 'gera um número aleatório entre 0 e 6, inteiro

  Select Case I

  Case 0

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\fire.ico")

  Case 1

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\snow.ico")

Case 2

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\earth.ico")

  Case 3

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\rain.ico")

  Case 4

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\water.ico")

  Case 5

    Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\sun.ico")

  Case 6

  Image1.Picture=LoadPicture("c:\arquiv~1\devstu~1\vb\graphics\icons\elements\litening.ico")

  End Select

End Sub

Obs.: Rnd gera valores aleatórios em ponto flutuante no intervalo de 0 a 1. Portanto, se quisermos valores, por exemplo, no intervalo de 200 a 400, basta fazermos “I=Rnd * 200 + 200”, onde I é uma variável inteira.

Isso decorre do fato de “I=Rnd*200”, gera valores aleatórios de 0 a 200. Somando-se 200, teremos valores de 200 a 400.

Se por exemplo quizermos gerar números entre 6 e 10:

a) Verifica-se a diferença do intervalo: 10 – 6 = 4;

b) Multiplica-se o aleatório pela diferença : y = RND * 4 (gera números entre 0 e 4);

c) Soma-se o valor mínimo : y = y + 6 , teremos valores entre 6 e 10.

Exercícios:

- Utilizando as propriedades Top e Left da Image, alterar o programa para que a Image apareça em locais aleatórios do formulário.

- Utilizando as propriedades Width e Height, alterar o programa para que a Image apareça em tamanhos diferentes (aleatórios).

- Atribuir à propriedade BackColor do formulário, uma cor aleatória ( Form1.BackColor=1000000*Rnd).

- Alterar o programa do foguete, onde o foguete se choca com o avião, introduzindo um valor aleatório ( Pode haver o choque ou não)

Sugestão: comparar o valor absoluto como distância mínima para o choque, fazendo algo mais ou menos assim:

           . . .

           aleatorio = Rnd * 13

           . . .

 If Abs(figfoguete.Top - figaviao.Top) < 300 And Abs(figfoguete.Left - figaviao.Left) < 300 Then

        i = 1

     Else

        figfoguete.Top = figfoguete.Top - aleatorio

        figaviao.Top = figaviao.Top + 10

        figfoguete.Left = figfoguete.Left + 7

        figaviao.Left = figaviao.Left + aleatorio

        figfoguete.Width = figfoguete.Width - 2

        figaviao.Width = figaviao.Width + 1

        figfoguete.Height = figfoguete.Height - 2

        figaviao.Height = figaviao.Height + 1

      End If

         . . .

GERAÇÃO DE PROGRAMAS EXECUTÁVEIS:

O VB possibilita a geração de programas executáveis ( .EXE ). Para isso, abra o item “File” do menu e selecione “Make NomeProjeto.EXE”. Direcione o programa para o drive desejado, forneça o nome para o programa executável e dê um click no botão Ok.

Grave em seu disquete a versão executável do programa de choque do avião com o foguete. Em seguida, saia do VB e execute o programa diretamente pelo “Executar” do Windows, ou pela linha do DOS.

Obs.: Os programas executáveis gerados pelo VB 6.0 podem ser processados independentemente do ambiente do VB, não necessitando de arquivos adicionais para sua execução (exceto, por um arquivo .DLL).

TRATAMENTO DE ERROS E DEPURAÇÃO DE PROGRAMAS

O VB oferece uma grande gama de ferramentas que possibilita o tratamento a erros. Erros podem surgir quando o programa está sendo desenvolvido ( o que é mais comum) ou posteriormente quando o programa está em fase de operação. Neste último caso os erros (ou BUGS) são geralmente imprevisíveis pois são consequência de falhas na lógica do programa, ocorrência de condições que não foram previstas, etc. Mesmo nestes casos, o VB oferece recursos que facilitam a detecção e correção de erros.

Vimos por exemplo a cláusula “On Error Goto label” no exercício do manuseio de arquivos. Essa cláusula em conjunto com as funções ERROR$() e “ERR” pode ser muito útil na depuração de programas e também no eventual tratamento a algum “bug” imprevisto. A pior coisa que pode acontecer a um programa é a ocorrência de um erro inesperado e irrecuperável, onde o operador do programa perca seu trabalho já realizado e fique sem informações sobre o que fazer e sem alternativas.

Os bons programadores prevêem os possíveis erros que podem ocorrer quando o programa estiver sendo executado por um usuário e procuram contornar os problemas que possam decorrer desses erros. A função ERR retorna um valor indicando o tipo de erro ocorrido, dando ao programador alguma possibilidade de contorná-lo. Alguns valores e significados:

Número do erro      Significado

11              Divisão por zero

13              Discordância de tipo de dado

61              Disco Cheio

64              Erro no nome do arquivo

A função ERROR$() retorna o significado do erro em inglês - exemplo :

print ERR

print ERROR$(ERR)

Exemplo:

Private Sub Form_Click()

   Dim X as single

   On error GoTo Exiberro

        x=1/0     ‘Divisão por zero!!!!

Exiberro:

    Print Err

    Print Error$(Err)

    Exit Sub

End Sub

Use a tecla F1 para obter maiores informações sobre essas funções.

Exercício: Modificar o programa acima para fazer a divisão por um valor fornecido pelo usuário. Caso ocorra erro, é possível ao usuário fazer três tentativas. Quando a divisão for válida, o programa mostra o resultado. Se continuar inválida após três tentativas, O programa exibe mensagem e encerra. Será algo parecido com o que segue:

...

Tenta = 0

On error goto erro

                ‘  cálculo da divisão

...

Exit Sub

Erro:

MsgBox (“... Erro na divisão  -  Tente novamente...”)

Tenta = tenta + 1

If tenta > 2 then 

              resume next ‘ instrução seguinte

else

             resume ‘ mesma instrução

endif

...

Talvez o uso de uma Input Box ajude.

Um outro erro muito comum cometido por um usuário em um ambiente multitarefa como o Windows, é a execução simultânea de um programa, mais de uma vez. Pode acontecer que o usuário esteja operando um determinado programa e faça uma pausa para realizar outra tarefa. Ao voltar, pode ser que esse usuário esqueça que o programa já está em execução e processe-o novamente. Na maioria dos casos, isso não causará nenhum problema. Porém, dependendo das características do processamento ou do programa, pode ser que se percam informações.

Nesse caso, o uso da propriedade “PrevInstance”, evitará que uma nova instância do programa ocorra. No exemplo abaixo, só poderá haver uma execução do programa por vez.

Option Explicit

Private Sub Command1_Click()

   Print "ola!!!"

   Print App.Path

End Sub

Private Sub Form_Load()

   If App.PrevInstance <> 0 Then

       MsgBox "Programa " + App.Title + "   Já está em execução!!!"

       End

   End If

End Sub

Quanto a erros de digitação, o próprio VB se encarrega de apontar os erros de sintaxe ocorridos em uma linha, quando passamos à linha seguinte. Para que isso aconteça, deverá estar ligada a opção “Auto Syntax Check”. Verifique se essa opção está ligada, no item de menu “Tools/Options”, na guia “Editor”.

Na fase de depuração e testes de um programa, vários erros podem ser localizados e consertados em tempo de execução, permitindo que se continue a executar o programa sem interrupção.

Para isso, verifique que esteja ligada a opção “Auto Data Tips”, também no item de menu “Tools/Options”, na guia “Editor”. Essa opção faz com que seja mostrado o conteúdo da variável onde o cursor está posicionado.

Vamos ver um pequeno exemplo de localização de erro, por meio de um programa (contendo um erro, a princípio) que exibe um label e um botão. 

Quando dermos clicks no botão, o caption do label deverá mostrar a quantidade de clicks que foram dados no botão.

Nossa tela será algo como abaixo:

Note que no código abaixo, declaramos a variável de nome “contador”, porém na atribuição do conteúdo dessa variável ao Caption do Label, escrevemos, erradamente, o nome da variável como contad0r.

Para que o nosso programa errado cumpra seus objetivos, verifique se na área “General” desse formulário, existe a cláusula “Option Explicit”, e se houver, elimine-a.

Essa cláusula obriga o programador a declarar todas as variáveis do programa, o que é bastante recomendável. Porém, em nosso exercício, necessitamos de um programa com erro, para investigarmos maneiras de encontrar erros

Se não tivermos ativado a opção de obrigatoriedade de declaração de variáveis, o VB irá auto declarar a variável “contad0r” e considerar “contador” e “contad0r” como duas variáveis distintas.

O código ( com erro ), será assim:

Private Sub CmdClicks_Click ()

   Static contador As Integer   ‘Variaveis static mantém o valor entre as chamadas.

   contador = contador + 1

   LblContaClicks.Caption = Str$(contad0r)

End Sub

Executando o programa, veremos que o mesmo não funciona a contento. Suponha que não sabemos porque, e portanto, vamos usar algumas ferramentas de debug para descobrir.

Em primeiro lugar vamos incluir um “Break Point” em algum local do programa onde desconfiamos que pode estar o erro.  Para isso, posicionamos o cursor na instrução desejada e pressionamos a tecla F9.

Em seguida, executamos o programa. Verifique que ao atingir a instrução onde você colocou o Break Point o processamento é interrompido e surge o programa na tela do micro. Aí basta posicionar o cursor sobre a variável ou selecionar a expressão que você quer avaliar e posicionar o cursor sobre a expressão selecionada. Após 1 segundo, surge a “hint” com o valor do conteúdo da expressão ou variável selecionada. Dessa maneira, você pode verificar se o valor apresentado corresponde ao esperado. 

A cada vez que você pressionar a tecla F8, o programa avança uma instrução para que se possa novamente verificar o conteúdo de variáveis ou valor de expressões. Ao chegar ao final da procedure, dê um click no botão Start para continuar a execução.

Você pode também incluir variáveis ou expressões na janela “Watches”, para acompanhar a evolução de seus conteúdos. Para isso, selecione a variável ou expressão desejada e abra os itens de menu “Debug/Add Watch”. Note que é aberta a janela “Watches”, com a variável ou a expressão selecionada sendo que seu valor vai variando dinamicamente, de acordo com o processamento do programa. 

Com o nosso exemplo de programa com erro acima, verificamos que o VB criou a variável contad0r, que não foi declarada explicitamente. Portanto em VB há a  possibilidade de uso de variáveis que não se necessita declarar. O benefício que isso pode trazer é um tanto duvidoso pois erros como o apontado acima são muito comuns de ocorrer, apesar do pequeno ganho de tempo com a não declaração de variáveis.

Se você achar que esse pequeno ganho de tempo não compensa o risco, obrigue-se a declarar todas as variáveis em seus programas, pois assim, uma variável com nome errado será imediatamente detectada pelo compilador do VB.

Para isso, antes de iniciar seu projeto, abra o item de menu Tools, subitem Options, escolha a guia Editor e selecione a opção REQUIRE VARIABLE DECLARATION. Isso fará com que, nos próximos projetos, seja incluída a cláusula “OPTION EXPLICIT” na área “GENERAL” do formulário.

Neste nosso exercício, como o compilador não incluiu a cláusula (ou nós a apagamos), você deverá digitar a cláusula “OPTION EXPLICIT” na área “GENERAL” do formulário que o compilador detectará as variáveis com nome errado que possam estar em seus programas.

Faça isso agora no programa “Conta Clicks” e verifique que o compilador denunciará a variável “contad0r”.

INSTRUÇÕES PARA CONTROLE DE MALHAS

Vamos fazer um programa bastante simples com o objetivo de mostrar as malhas existentes em VB e a forma que as mesmas possuem.

Primeiramente, faremos o programa usando o objeto “Botão” e em seguida modificaremos os botões para controles na forma de “Itens de Menu”.

A janela de nossa aplicação será mais ou menos assim:

                      [image: image38.png]W CONTROLES [Tl

DADO MENSAGEM
[ itrenche | selct case

whie wend doloop while





Obs.: Em aplicações Windows, quando alguma operação vai levar algum tempo, é padrão deixar o apontador do mouse na forma de uma ampulheta, enquanto durar essa operação.

No código abaixo, você vai ver um exemplo disso. No Windows 95 há possibilidade de escolhermos uma ampulheta animada. Para isso, escolher “Ampulhetas Animadas”com o seguinte caminho: “Configuração / Painel de Controle / Mouse / Ponteiros / Ampulhetas Animadas”.

O código associado aos botões é o seguinte:

Option Explicit

Private Sub cmdif_Click ()     ‘FORMA DO COMANDO “IF-THEN-ELSE”

      If Val(txtdado.text) < 10 Then

          txtmensagem = "Valor menor  que 10"

      Else

          txtmensagem = "Valor  maior que 10"

      End If

End Sub

Private Sub cmdloop_Click ()   ‘FORMA DO “DO...LOOP”

     mousepointer = 11      ‘ENQUANTO DURAR O LOOP, O PONTEIRO

     Dim n As Integer          ‘FICA NA FORMA DE AMPULHETA.

     n = Val(txtdado.text)

     Do

         n = n - 1

         txtmensagem.Text = n

     Loop While n > 0

     mousepointer = 0      ‘O PONTEIRO VOLTA À FORMA NORMAL.

End Sub

Private Sub cmdselect_Click ()   ‘FORMA DO “SELECT CASE”

     Select Case Val(txtdado.Text)

         Case 0 To 10

             txtmensagem.Text = "Valor entre 0  e 10"

         Case 11 To 100

             txtmensagem.Text = "Valor entre 11 e 100"

         Case Else

             txtmensagem.Text = "Outros valores"

     End Select

End Sub

Private Sub cmdwhile_Click ()   ‘FORMA DO “WHILE... WEND” 

     mousepointer = 11

     Dim n As Integer

     n = Val(txtdado.text)

     While n > 0

         n = n - 1

         txtmensagem.Text = n

     Wend

     mousepointer = 0

End Sub

Private Sub Form_DblClick ()   ‘ SE FOR DADO UM DUPLO CLICK NO 

      PrintForm                 ‘FORMULÁRIO, É EXECUTADO O “PrintForm”.

End Sub                          ‘EXPERIMENTE...

Teste o programa para verificar como funciona. Em seguida, modifique-o para eliminar os botões, fazendo o programa funcionar por meio de menus, onde haverá um item principal de menu “COMANDOS”, com os sub tens “IF THEN ELSE”,  “SELECT CASE”, WHILE WEND”, “DO LOOP WHILE”.

Para isso, existe na Barra de Menu do VB, no item Tools, o sub item Menu Editor que nos permite criar menus em nossa aplicação. Para isso, abra a janela “Menu Editor” e explore-a. 

Obs.Para habilitar o item Menu Editor, basta dar um click no formulário para selecioná-lo. 

Alguns itens importantes no momento, são iguais aos utilizados em um botão:

Caption: É o título que aparecerá no menu.

Name: É o nome do item de menu, onde será feita a programação.

Shortcut: Inclusão de uma tecla de atalho, como por exemplo, “Ctrl + I”.

(,(,(,(  : Pressionando essas teclas, iremos “ajustando” os itens de menus e sub menus.

Next, Insert, Delete: São usados para mover ao próximo item de menu, inserir um novo item e apagar o item selecionado.

Obs.: É muito comum a utilização de agrupamentos de alguns sub itens do menu, onde esses sub itens ficam separados dos demais por meio de uma linha. Para fazer a separação, coloque o sinal de menos ( - ) no Caption de um item de separação, que o VB providenciará uma linha separando os itens. 

Faça alterações para que nossa janela fique com a seguinte forma:

Em seguida, vamos transportar o código de cada botão, para o item de menu correspondente, utilizando o recurso do Windows de recortar ( CTRL X) e colar ( CTRL V)o código dos botões para os sub itens de menu. Assim, nosso programa final ficará assim:

Private Sub menudoloop_Click ()   ‘FORMA DO “DO...LOOP”

     mousepointer = 11

     Dim n As Integer

     n = Val(txtdado.text)

     Do

         n = n - 1

         txtmensagem.Text = n

     Loop While n > 0

     mousepointer = 0

End Sub

Private Sub menuif_Click ()     ‘FORMA DO COMANDO “IF-THEN-ELSE”

      If Val(txtdado.text) < 10 Then

          txtmensagem = "Valor menor  que 10"

      Else

          txtmensagem = "Valor  maior que 10"

      End If

End Sub

Private Sub menuselect_Click ()   ‘FORMA DO “SELECT CASE”

     Select Case Val(txtdado.Text)

         Case 0 To 10

             txtmensagem.Text = "Valor entre 0  e 10"

         Case 11 To 100

             txtmensagem.Text = "Valor entre 11 e 100"

         Case Else

             txtmensagem.Text = "Outros valores"

     End Select

End Sub

Private Sub menuwhile_Click ()‘FORMA DO “WHILE... WEND” 

     mousepointer = 11

     Dim n As Integer

     n = Val(txtdado.text)

     While n > 0

         n = n - 1

         txtmensagem.Text = n

     Wend

     mousepointer = 0

End Sub

Finalmente, remova  os botões da aplicação pois não são mais necessários.

USO DA ÁREA DE TRANSFERÊNCIA

Em nosso próximo programa vamos utilizar o objeto “CLIPBOARD” (Área de Transferência) do Windows. Em VB é possível utilizar a área de transferência, manipulando textos ou gráficos em nossos programas. A sintaxe é a seguinte:

Método

Efeito Correspondente
Clear


Limpa a Área de Transferência

GetText

Pega e traz o texto que está na Área de transferência

GetData

Pega e traz os dados gráficos da Área de Transferência

SetData

Transfere dados gráficos para a Área de Transferência

SetText

Transfere texto para a Área de Transferência

Como trabalho inicial vamos construir uma tela bem simples, que deverá parecer como abaixo:

A caixa de texto de nome TxtUserArea, tem as propriedades MultiLine=true e ScrolBars=Vertical.

Utilizando-se o conhecimento acima, o código de nosso programa ficará assim:

Option Explicit

Private Sub cmdcolar_Click ()

            txtuserArea.SelText = clipboard.GetText

End Sub

Private Sub cmdcopiar_Click ()

            clipboard.Clear

            clipboard.SetText txtuserArea.SelText

End Sub

Private Sub cmdrecortar_Click ()

           clipboard.Clear

           clipboard.SetText txtuserArea.SelText

           txtuserArea.SelText = " "

End Sub

Private Sub cmdsair_Click ()

            End

End Sub

Vamos converter nossos botões COPIAR, RECORTAR, COLAR em um item de menu “EDITAR”, que deverá ficar na seguinte forma:




EDITAR




    COPIAR




    RECORTAR




    COLAR

Após criado o menu “EDITAR” e os sub menus “COPIAR”, “RECORTAR” E “COLAR”, transfira para os itens dos sub menus o código em VB. contido nos botões correspondentes, usando o recurso do Windows de RECORTAR (Ctrl X) e COLAR (Ctrl V). 

Após feito isso, faça o teste para verificar se o programa está bom e elimine os botões de sua aplicação, deixando apenas o menu “EDITAR” e o botão “SAIR”.
Como exercício, incluir na aplicação mais dois itens de menu: “Arquivo” e “Formatar”. No item “Arquivo”, colocar  sub itens para Gravar, Ler Arquivos e Sair ( transfira o botão SAIR). 

Obs 1. Você já fez um programa que possuía formulários para leitura e gravação de arquivos, há mais ou menos duas aulas atrás. Utilize aqueles mesmos formulários, acrescentando-os a este programa, por meio do item de menu “Project” sub-item “Add Form”, selecione a guia “Existing” e forneça o caminho onde se encontram os formulários desejados.

Obs 2. Caso ocorra o erro “Errors during load...”, é muito provável que você esteja tentando adicionar um formulário que possui o nome interno (propriedade NAME) igual ao nome de um formulário já existente no projeto. Isso é muito comum, pois programadores inexperientes costumam manter inalterados os nomes Form1, Form2, etc.

Para contornar esse problema, mude os nomes dos formulários de seu projeto, para nomes personalizados e mais sugestivos.

Para visualizar os formulários que você acabou de incluir ao projeto, dê um click nos itens de menu “View/Project Explorer”, e dê um click duplo no nome do formulário que você quer visualizar. 

No item “Formatar” inclua sub itens para mudar o tipo de letra e tamanho do fonte.

Exemplo:

Private Sub MENU10_Click ()

    txtuserArea.FontSize = 10 ‘ Faz o tamanho do fonte igual a 10 pixéis.

End Sub

Private Sub MENU15_Click ()

    txtuserArea.FontSize = 15

End Sub

Private Sub MENUITALICO_Click ()

   txtuserArea.FontItalic = Not txtuserArea.FontItalic ‘ Faz o fonte itálico e desfaz.

End Sub

Private Sub MENUNEGRITO_Click ()

    txtuserArea.FontBold = Not txtuserArea.FontBold ‘Faz o fonte negrito e desfaz.

End Sub

Menu PopUp (ou menu instantâneo): Incluir mais um item de menu com dois sub-itens deixá-lo com a propriedade Visible=false. Se o Name desse item for Mnupop, ele poderá ser chamado pelo método PopUpMenu, quando se dá um click no botão “Sair”, com a tecla direita do mouse.

Private Sub CmdSair_MouseDown(Button As Integer, Shift As Integer, X As Single, Y As Single)

   If Button = vbRightButton Then

        FrmBlocoDeNotas.PopupMenu   Mnupop    ‘ Nome do item de menu que está com Visible=false

   End If

End Sub

Private Sub mnuitem1_Click()

   txtuserArea.BackColor = vbRed

End Sub

Private Sub mnuitem2_Click()

   txtuserArea.ForeColor = vbYellow

End Sub

No final, seu menu deverá ficar assim:

USO DE UM FORMULÁRIO MDI (Multiple Document Interface)
Veremos agora como fazer a integração de um sistema, onde se possui uma janela composta de vários ícones, sendo que por meio de cada um deles pode-se executar um módulo do sistema. Isso é feito por meio de um formulário chamado MDI.

Afigura abaixo nos mostra um formulário MDI, com ícones para execução de programas. Nosso próximo exercício será construir uma aplicação como a abaixo.

Para isso, vamos abrir o item de menu “Project” e escolher “ Add MDI Form”.

Só é possível haver um formulário MDI para cada aplicação.

Incluir no formulário MDI as pictures Picture1 e Picture2 de acordo com o layout abaixo.

Inserir na Picture1, 3 objetos “Image”, sendo Image1, Image2, Image3, que serão objetos de uma “Barra de Ferramentas”. Na propriedade “Picture”de cada uma das “Images” inserir um ícone a seu gosto, sendo que em nosso exemplo foram:

- c:\arquiv~1\devstu~1\vb\graphics\icons\COMPUTER\MAC02.ICO

- c:\arquiv~1\devstu~1\vb\graphics\icons\OFFICE\RULERS.ICO

- c:\arquiv~1\devstu~1\vb\graphics\icons\MISC\CLOCK03.ICO

Obs1.: O bom senso e a Semiótica nos dizem que um ícone deve lembrar a ação que ele representa.

Obs2.: Ao invés de utilizarmos os objetos “Image”, poderíamos utilizar botões especiais do grupo “Sheridan 3D Control”. Porém como iremos utilizar esses botões em um outro exercício posteriormente, ficaremos aqui com o objeto “Image”.

Setar a propriedade WindowState = 2. Assim, quando executado, o formulário ocupará a tela toda.

Inserir na propriedade “Picture”da Picture2 um desenho para preencher a tela.

Em nosso exemplo, inserimos a figura de um computador, que está no local : c:\arquiv~1\devstu~1\vb\graphics\metafile\COMPUTER.WMF.

Aqui deve ser um arquivo tipo  .WMF (Windows Meta File) para ajustar-se ao tamanho da Picture, ficando como na figura abaixo:.

Está pronta a interface do formulário MDI. Agora vamos incorporar ao projeto os programas que irão compor o sistema. Nosso sistema será composto pelo formulário MDI, que conterá três programas distintos: o programa “CONTROLES”,que acabamos de fazer e mais dois outros programas, como por exemplo a calculadora e o relógio despertador.

Poderíamos ir desenvolvendo programas e ir incorporando-os ao projeto ou simplesmente usar programas que já temos prontos. Para isso, vamos carregar os módulos (arquivos com extensão .FRM) que já temos prontos em disquete.

Passos a seguir:

-Abrir o item  de menu “Project” e escolher “Add Form”. Selecione a guia “Existing” e, acrescente um a um, os formulários (arquivo .FRM) dos programas “CONTROLES”, “CALCULADORA” e “DESPERTADOR”. Isso é feito pelo nome que você atribuiu ao formulário quando você o gravou em disquete.

-Para cada um dos formulários, setar para “TRUE” a propriedade “MdiChild.

-A propriedade “Name” de cada formulário deve ser diferente, pois é pelo “Name” que os programas são executados.

Atenção:

Quando se executa um programa em VB, a execução inicia pelo primeiro formulário criado pelo próprio VB (Form1). Se quisermos que a execução inicie por um formulário diferente desse,  devemos abrir o item “Project” do menu, selecionar o sub item “Project1 Properties”, escolher a guia “General” abrir a lista “Startup Object” e dar um click no nome do formulário pelo qual queremos que seja iniciada a execução, que neste caso é o MDIForm1.

Nosso código em VB ficará assim:

Option Explicit

Private Sub Image1_Click ()

     picture2.Height = 0       ‘COMPRIME A ÁREA DE USUÁRIO DO MDI

                                           ‘PARA CABER O FORMULÁRIO “CONTROLES”.

     LoopForm.Show   ‘ CARREGA E EXECUTA O PROGRAMA “CONTROLES”

End Sub

Private Sub Image2_Click ()

    picture2.Height = 0   ‘COMPRIME A ÁREA DO MDI

    CalcForm.Show‘ CARREGA E EXECUTA O PROGRAMA “CALCULADORA”

End Sub

Private Sub Image3_Click ()

      picture2.Height = 0 ‘COMPRIME A ÁREA DO MDI

      AlarmForm.Show‘ CARREGA E EXECUTA O PROGRAMA “DESPERTADOR”

End Sub

OBS.: Se o aplicativo “filho”que estiver sendo executado possuir menus, esses menus aparecerão no fomulário MDI.

Exercício:
Criar o menu “JANELA” no formulário MDI para selecionar o tipo de arranjo que queremos para os aplicativos “filhos”do MDI. Para isso, seguir os seguintes passos:

- Selecionar o Formulário MDI - para isso, é necessário dar um click no formulário MDI (ou seja, no “Grid”, que é a parte pontilhada do formulário);

- Abrir o item Tools do menu;

- Abrir o “Menu Editor”;

-Ativar o item “  Window List”;

-Criar o menu...:

 JANELA 

CASCATA

LADO A LADO HORIZONTAL

LADO A LADO VERTICAL

-Digitar o seguinte código;

Private Sub MenuCascata_Click()

       MdiForm1.Arrange 0     ‘  ARRANJA EM CASCATA

End Sub

Private Sub MenuLLHorizontal_Click()

       MdiForm1.Arrange 1     ‘   ARRANJA EM LADO A LADO HORIZONTAL

End Sub

Private Sub MenuLLVertical_Click()

       MdiForm1.Arrange 2     ‘   ARRANJA EM LADO A LADO VERTICAL

End Sub

Mais exercícios:

1 - Substituir os controles “Image” por botões especiais. Vamos utilizar um tipo especial de botão que pode conter uma figura gráfica, além de seu Caption. Para isso, vamos incluir um novo conjunto de controles em nossa “ToolBox”. Acione o item de menu Project/Components e inclua o grupo “Sheridan 3D Controls”. Após feito isso, troque as “Images” por botões do tipo “SSCommand”.

A figura gráfica que deverá aparecer junto com o Caption do botão é um ícone que você irá inserir na propriedade Picture do botão. 

Em seguida, abra o item General/Declarations e copie o código das Images, para o evento Click dos novos botões.

2 - Substituir a Picture1 (Ou seja, a Barra de Ferramentas de sua aplicação), pelo controle TooBar, descrito a seguir.

CONTROLE TOOLBAR.

O controle ToolBar poderá ser utilizado ao selecionarmos o item de menu Project/Components e incluirmos em nossa ToolBox o objeto “Microsoft Windows Common Controls 5.0”.

Ao ser acionado esse item, diversos componentes ficam disponíveis, estando entre eles, a ToolBar e a ImageList.

Descrição:

ImageList: Controle que possui a propriedade Custom que é de nosso intesse. Ao ser acionada, nos oferece a opção Images:

- InserPicture: Botão que insere uma nova figura na lista;

- Index: É a propriedade que identifica uma determinada imagem pelo seu índice na lista;

- RemovePicture: Botão que remove a imagem da lista.

ToolBar: Esse controle também possui a propriedade Custom que ao ser acionada nos oferece as seguintes opções:

- General:

· ImageList: Determina em qual ImageList as figuras do Toolbar estão armazenadas;

· MousePointer: Determina o tipo de cursor do mouse que será mostrado ao passar com o apontador sobre a Toolbar;

· ButtonWidth e ButtonHeight: Determina o tamanho dos botões da Toolbar.

-Button:

· InsertButton: Clique para inserir um novo botão;

· Index: Determina a ordem na qual os botões serão apresentados;

· Caption: Texto que ficará abaixo de cada um dos botões;

· Description: Descrição do botão;

· Key: String que identificará cada um dos botões;

· Value: Determina se o botão estará pressionado ou não ao iniciar a aplicação;

· Style: Determina o tipo de botão:

1. DefaultButton: Botão comum da Toolbar. Ao click do mouse ele aperta e ao soltar ele levanta;

2. Check: Semelhante ao CheckBox do Visual Basic;

3. ButtonGroup: Possibilita apenas um dos botões do grupo selecionado - semelhante ao OptionButton do Visual Basic;

4. Separator: Deixa um espaço padrão entre os botões;

5. PlaceHolder: Deixa um espaço padrão entre os botões, de tamanho definido na propriedade Width.

· Tag: Propriedade auxiliar;

· ToolTip Text: Texto que aparece quando o apontador do mouse está sobre o botão ( Utilizada para descrever a “HINT” que será exibida);
· Image: Índice da imagem do botão, de acordo com a propriedade Index da ImageList.

UTILIZAÇÃO:

Inicialmente devemos criar um componente ImageList que armazenará as figuras que compõem a nossa barra de ferramentas. Para isso, devemos proceder da seguinte maneira:

1. Inserimos um componente ImageList no formulário (Atenção... Se você a incluir sobre um controle Picture, ela ficará oculta sob a Picture).

2. Entramos na sua propriedade Custom, selecionamos a guia General e selecionamos o tamanho desejado ( experimente 32 x 32);

3. Selecione a guia Images;

4. Click em Insert Picture;

5. Escolha o arquivo .ICO ou .BMP que será lido e numere a figura na caixa Index (ou mantenha o número que o VB colocou).

Uma vez inseridas todas a figuras, devemos inserir o componente ToolBar na aplicação. Daí:

1. Inserimos o ToolBar, dando um duplo click no ícone correspondente e arrastando-a para a posição desejada (superior ou inferior);

2. Dê um click duplo na propriedade Custom e selecione a ficha General;

3. No quadro Combo “ImageList”, selecione a ImageList1;

4. Clique na ficha Buttons e dê um click em InsertButton, para inserir um botão;

5. Preencha os campos Index, Key e Image com os dados conforme a descrição do controle ImageList:

· Index: Ordem em que o botão é apresentado;

· Key: String que você deverá digitar, que vai identificar o botão. Tente os strings “Loops, Calcula, Desperta”; sendo um em cada botão.

· Image: Índice da figura que será apresentada no botão, de acordo com o Índex da ImageList.

6. Coloque no item “ToolTipText” de cada um dos botões, uma breve explicação ( “Hint” ) do que faz o botão (certifique-se que a check box “Show Tips” da guia General esteja ligada). 

Em seguida, devemos estabelecer o código para cada um dos botões. Dê um duplo click na barra de ferramentas que você acabou de criar para abrir a janela de código do VB.

Utilizamos o Select Case da seguinte maneira:

A variável Button guarda o botão que foi pressionado. Identificaremos esse botão pela propriedade Key, que é única para cada botão. 

Exemplo:

Private Sub Toolbar1_ButtonClick(ByVal Button As Button)

Select Case Button.Key

       Case “Loops”

                                    ‘Codigo do botão Loops

                  Picture2.Height = 0

                  LoopForm.Show

       Case “Calcula”

                                     ‘Codigo do botão Calcula

                  Picture2.Height = 0

                  CalcForm.Show

       Case “Desperta”

                                      ‘Codigo do botão Desperta

                  Picture2.Height = 0

                  AlarmForm.Show

End Select

End Sub

Exercício: Inclua mais um botão, que ao receber o click do mouse, aciona o programa “Bloco de Notas”, que você fez no exercício anterior.

Obs. É possível incluir na aplicação, várias barras de ferramentas.

MANUSEIO GRÁFICO COM VB:
O VB permite o desenho em objetos “Form” e “Picture Box”, podendo-se desenhar pontos, linhas, retângulos, círculos e outros tipos de figuras derivadas destas.

O sistema de coordenadas para os gráficos inicia em (0,0), aumentando da esquerda para a direita e de cima para baixo.

A espessura das linhas gráficas é definida pela propriedade “DrawWidth”, podendo variar de 1 a 32767 pixels, e a cor pode ser definida de três maneiras: 


pelo uso da função RGB;


pelo uso da função Qbcolor;


pelo fornecimento direto do valor da cor em hexadecimal.

Desenho de pontos: Função “Pset(x,y), cor”.

X e Y são a abscissa e ordenada de onde se deseja o ponto, e “cor” é uma função ou o proprio valor da cor desejada. Veja o exemplo no programa abaixo:

                  Private Sub Form_Click ()

                         PSet (Scalewidth / 2, Scaleheight / 2), &HFF&

                  End Sub

O programa acima, insere um ponto vermelho no centro do formulário, quando for dado um click em qualquer lugar do formulário.

Verifique o uso das funções RGB e Qbcolor com a tecla F1.

Se quizermos desenhar o ponto em uma “Picture Box”, basta direcioná-lo para esse objeto.

Inclua a “Picture1” no formulário, de acordo com o layout e programação abaixo:

Private Sub Form_Click ()

    Dim I As Integer

    For I = 1 To 255 Step 1      ' Desenha uma linha colorida.

          PSet (scalewidth / I, ScaleHeight / I), RGB(I, 0, 255 - I)

    Next I

End Sub

Private Sub Picture1_Click ()

     Dim I As Integer

     For I = 1 To 255 Step 1

         Picture1.PSet (Picture1.ScaleWidth / 255 * I, Picture1.ScaleHeight / 2), RGB(255 - I, 0, I)

     Next I

End Sub

Execute o programa e dê um click na Picture1. Note que os pontos formarão uma linha de pontos que vão mudando de azul para vermelho, à medida que são mudados os argumentos da função RGB.

Se você der um click no formulário, obterá o mesmo efeito, só que a linha de pontos será exibida na diagonal do formulário.

DESENHO DE LINHAS:

É feito por meio da função LINE que tem a seguinte sintaxe:

“Objeto.Line (x1, y1) - (x2, Y2), cor” onde x1, y1 são as coordenadas do primeiro ponto, e x2,y2 são as coordenadas do segundo ponto.

Acompanhe o exemplo a seguir :

Private Sub Form_Click ()      ‘LINHA VERMELHA NO FORMULÁRIO

       Dim i As Integer

       For i = 1 To 9

            drawwidth = i    ‘VAI AUMENTANDO O TAMANHO DA LINHA

            Line (0, 0)-(scalewidth, scaleheight), &HFF&

       Next i

End Sub

Private Sub Picture1_Click ()   ‘LINHA VERDE NA PICTURE BOX

       Dim i As Integer

       For i = 1 To 9

            picture1.DrawWidth = i

            picture1.Line (0, 0)-(picture1.ScaleWidth, picture1.ScaleHeight), &HFF00&

       Next i

End Sub

Podemos também definir o tipo da linha com a propriedade “DRAWSTYLE”, que varia de 0 a 6. Verifique esses valores com o Help on Line (F1).

Exercício: Fazer um programa que trace uma linha entre dois pontos onde se deu o click do mouse. Usar a “Sub MouseDown” para obter as coordenadas dos pontos.

A janela  e o programa será algo parecido com o abaixo:

Dim xant, yant As Single  ‘X anterior e Y anterior

Private Sub Form_MouseDown (Button As Integer, Shift As Integer, x As Single, y As Single)

       If xant + yant = 0 Then     ‘ESTE E O PRIMEIRO CLICK( PRIMEIRO PONTO)

            xant = x

            yant = y

      Else                                    ‘SEGUNDO CLICK (SEGUNDO PONTO)

            Line (xant, yant)-(x, y)   ‘IMPRIME A LINHA ENTRE OS PONTOS

            xant = 0           ‘ZERA COORDENADAS P/ PROXIMA LINHA.

            yant = 0

      End If

End Sub

Vamos incrementar um pouco mais nosso programa, incluindo um menu onde se pode escolher o estilo da linha, sua espessura e cor.

Verifique com a tecla F1 os valores correspondendes às propriedades Drawstyle (estilo da linha) e DrawMode (estilo do desenho).

Acione o “Menu Editor” e insira um menu em seu programa, mais ou menos como o abaixo:

Dessa maneira, sua aplicação ficará assim:

Dim xant, yant As Single

Dim COR As Long

Sub Form_MouseDown (Button As Integer, Shift As Integer, x As Single, y As Single)

If xant + yant = 0 Then

   xant = x

   yant = y

Else

   Line (xant, yant)-(x, y), COR

   xant = 0

   yant = 0

End If

End Sub

Private Sub menazul_Click ()

    COR = &HFF0000

End Sub

Private Sub mengrossa_Click ()

    drawwidth = 9

End Sub

Private Sub meninverso_Click ()

     drawmode = 4

End Sub

Private Sub meninvertela_Click ()

    drawmode = 6

End Sub

Private Sub menmedia_Click ()

    drawwidth = 5

End Sub

Private Sub menormal_Click ()

     drawmode = 13

End Sub

Private Sub menpreta_Click ()

   COR = &H0&

End Sub

Private Sub menufina_Click ()

    drawwidth = 1

End Sub

Private Sub menverde_Click ()

    COR = &HFF00&

End Sub

Private Sub menvermelha_Click ()

    COR = &HFF&

End Sub

DESENHO DE RETÂNGULOS:

Também é feito pela função Line, com a seguinte sintaxe: “Objeto.Line (x1, y1) - (x2, Y2), cor, B”. Se “B” for especificado, significa que ao invés de uma linha será impresso um “Box” entre as coordenadas fornecidas.

Se fornecermos a propriedade “Fillstyle”do objeto, podemos preencher os retângulos com vários padrões. O default de Fillstyle é transparente representado pelo valor 1. Verifique com a tecla F1 os valores de “Fillstyle”.

No programa exemplo acima, inclua o parâmetro “B” (Box) na chamada da função “Line” e inclua um novo item no menu, de nome “Achura”, onde os se atribua os valores de “Fillstyle”: Fillstyle=0  (sólido), Fillstyle=1  (trasparente), Fillstyle=2 (horizontal) e Fillstyle=3  (vertical).

Exemplo: 

Private Sub menvertical_Click()

     fillstyle = 3

End Sub

DESENHO DE CÍRCULOS:

É feito por meio da função CIRCLE que tem a seguinte sintaxe:

“Objeto.Circle (x, y), Raio, Cor, Inicio,Fim, Aspecto” onde “(x, y)” são as coordenadas do centro do círculo, “Raio” é o raio do círculo em Twips, “Início e Fim” são valores em radianos, que vão de zero a 2(, indicando que queremos desenhar um semi círculo e “Aspecto” é o aspecto que queremos da figura, ou seja, é um valor que significa  a razão vertical para horizontal da elipse.

Veja o exemplo abaixo:

Dim INICIO, FIM, ASPECTO As Single

Private Sub Form_MouseDown (Button As Integer, Shift As Integer, x As Single, y As Single)

     Dim INICIO, FIM, ASPECTO As Single

     Dim raio As Integer

    raio = Val(txtraio.Text)

     INICIO = Val(txtinicio.Text)

     FIM = Val(txtfim.Text)

     ASPECTO = Val(txtaspecto.Text)

     Circle (x, y), raio, &HFF, INICIO, FIM, ASPECTO

End Sub

Obs.: Os parâmetros drawmode, drawstyle, fillstyle, etc., valem também para o método “Circle”.

TROCA DE INFORMAÇÕES COM OLE

OLE (Object Linking and Embedding) : É a técnica utilizada pelo WINDOWS  que nos -permite “colar” ou embutir objetos em um controle OLE.

É possível inserir um objeto no controle OLE em “tempo de projeto” ou em “tempo de execução”.

Para verificar os tipos de objetos possíveis de se colar em um controle OLE, analise a tela “Inserir Objeto”. Essa tela é obtida quando se inclui o controle OLE no formulário.

Vamos fazer um programa para estudar o controle OLE. Crie um projeto novo e inclua no formulário o objeto OLE. Sua tela deverá parecer como abaixo:

Podemos colar por exemplo, um gráfico ou uma planilha feitos em excel, no controle OLE de nosso formulário. Após analisar a tela acima, dê um click em “Cancelar”.

Vamos visualizar novamente a tela “Inserir Objeto”. Posicione o apontador do mouse sobre o objeto OLE e dê um click com o botão direito do mouse. Selecione o item “Insert Object”.

Escolha “Grafico do Microsoft Excel 5.0” e verifique que um gráfico é inserido no controle OLE de seu formulário.

Execute o programa e dê um click duplo no controle OLE. Verifique que você pode fazer alterações nos dados da planilha e estas serão refletidas nas colunas do gráfico e vice versa.

Vamos ver como fazer para manusear uma planilha e um gráfico que montamos com dados criados por nós.

Pare o programa em VB e entre no Excel.    Crie a planilha de nome “Plaminha.xls”, com as células a1,b1,c1, contendo os valores 10, 20, 30, respectivamente.

A seguir, faça o gráfico de colunas parecido com o abaixo.

Para fazer o gráfico, seguir os seguintes passos:

-Introduza os dados na planilha;

-Selecione os dados da planilha;

-Abra o item “Inserir” do menu e escolha “Gráfico”;

-Selecione “Como Nova Planilha”;

-Escolha como tipo de gráfico “Colunas”;

-Salve a planilha como “A:Plaminha”.

Em seguida , volte ao VB, selecione o controle OLE, faça a propriedade “Sizemode” = “Stretch”, abra a propriedade “SourceItem” e na janela “Inserir Objeto”, colocar no item Arquivo : “A:Plaminha.Xls”.

Execute o programa e dê um click duplo no objeto OLE. Verifique que é possível manusear a planilha e o gráfico que você criou. As alterações que se fizer em um deles refletem-se no outro e vice-versa.

Obs.: Se você deu um click duplo no gráfico e deseja retornar o controle ao programa em VB, pressione a tecla “Esc”.

Podemos também gravar os novos dados após termos feito alterações em nossa planilha, assim como podemos também ler novamente os dados da planilha em um disco. Acompanhe o programa abaixo:

Private Sub CmdSalvar_Click()

   Open "b:test.ole" For Binary As #1

   OLE1.SaveToFile 1

   Close #1

End Sub

Private Sub CmdLer_Click()

   Open "b:test.ole" For Binary As #1

   OLE1.ReadFromFile 1

   Close #1

End Sub

Obs.: Para começar com o arquivo que você quer iniciar, coloque os comandos de abertura e leitura no evento “Form Load”. 

No programa acima usamos os métodos SaveToFile e ReadFromFile. Utilize o Help On Line (F1) para saber mais sobre esses métodos.

Os arquivos manuseados pelos métodos SaveToFile e ReadFromFile devem ser abertos no modo binário. Por isso, após a gravação do arquivo, não é mais possível abrí-lo com o Microsoft Excel.

Execute o programa acima e verifique que agora é possível salvar a planilha e seu gráfico, após realizar alterações nos dados.

Exercício: Incluir no programa acima, uma caixa de texto onde se digita o nome do arquivo que se vai ler ou gravar. O formulário deverá ficar mais ou menos como abaixo, com a seguinte sugestão de código e especificações:

Faça “Visible=False” para o Label e a caixa de texto com o nome do arquivo;

Faça os botões de leitura e gravação com “Enabled=false”.

Sugestão de programa:

Private Sub cmdcancela_Click()

   Label1.Visible = False

   txtnomarq.Visible = False

   cmdler.Enabled = False

   cmdsalvar.Enabled = False

   cmdlersalvar.Enabled = True

End Sub

Private Sub cmdler_Click()

   Open txtnomarq For Binary As #1

   OLE1.ReadFromFile 1

   Close #1

End Sub

Private Sub cmdlersalvar_Click()

   Label1.Visible = True

   txtnomarq.Visible = True

   cmdler.Enabled = True

   cmdsalvar.Enabled = True

   cmdlersalvar.Enabled = False

   txtnomarq.SetFocus

End Sub

Private Sub cmdsalvar_Click()

   Open txtnomarq For Binary As #1

   OLE1.SaveToFile 1

   Close #1

End Sub

Você pode preferir que seu programa inicie sem nenhum arquivo, pois você pretende fazer a leitura posteriormente.

Para isso, selecione o controle OLE, abra as opções da propriedade “Class” e escolha a classe “Excel Chart 5”.

Execute o programa e faça a leitura do arquivo de sua preferência ( lembre-se que o arquivo deve ser binário).

Note que a entrada do programa dessa maneira, fica muito mais rápida.

Se você quizer que a aplicação “OLE Server” apareça com o menu original em sua aplicação VB, você deve incluir pelo menos um item de menu na sua aplicação em VB e deixá-la com a propriedade “Visible=False”. 

Usando o menu “Editar-Copiar/Colar” da aplicação “OLE Server”, você tem uma boa maneira para transitar todos os tipos de dados entre a aplicação VB usando a Área de Transferência: Clipboard e os métodosGetText (copiar texto), SetText (colar texto), Setdata (copiar dados gráficos) e GetData (colar dados gráficos).

Exercício:

Faça um teste com outros tipos de arquivos tais como o PAINTBRUSH ou o WORD. Para isso, dê um click com o botão direito  do mouse sobre o controle OLE, dê um click em “Insert Object”, selecione o botão de opção “Criar do Arquivo”, click no botão “Procurar”, escolha o arquivo desejado e click em “Inserir”. Após você der o click em “OK”, o VB fará automaticamente toda a configuração.

Observação: Como estamos trabalhando o controle OLE em tempo de projeto, uma vez sendo selecionado o tipo de objeto, este não poderá ser alterado. Para que se possa alterar o tipo de objeto em tempo de execução, veja o próximo tópico da apostila.

CRIAÇÃO E INSERÇÃO DE OBJETOS EM TEMPO DE EXECUÇÃO.

Neste caso, as propriedades e métodos são utilizados em tempo de execução. Exitem vários métodos e propriedades utilizados para isso. Vamos ver apenas uma maneira simples onde o operador pode escolher o tipo de objeto que vai inserir no controle OLE.

Isso pode ser feito chamando-se durante a execução, a janela “Insert Object” e selecionando-se o tipo de objeto a ser contido no controle OLE. Isso é feito pelo método “InsertObjDlg”

Exercício: Inclua um botão “Inserir Objeto” na sua aplicação e coloque no evento Click desse botão a instrução “Ole1. InsertObjDlg”.

Execute o programa e selecione tipos diferentes de objeto.

CONTROLE “CommonDialog”: É utilizado para facilitar a execução de várias tarefas que normalmente um programa em Windows executa, tais como abrir um arquivo, salvar e imprimir dados, selecionar cores e selecionar fontes.

Vamos verificar como funciona esse controle. Para isso, vamos abrir um projeto novo e incluir o “CommonDialog Control” ao nosso projeto. Abra o item de menu “Project” dê um click em “Components”, selecione a guia “Controls”, dê um click na CheckBox do controle “Microsoft Common Dialog Control 5.0” e pressione o botão “OK”.

Note que em sua Caixa de Ferramentas foi incorporado o ícone  [image: image53.png]


 . Inclua-o em seu formulário. Note que não é possível aumentá-lo de tamanho, pois ele não fica visível em tempo de processamento.

Selecione o CommonDialog Control e tecle F1 para obter algumas informações.

Os seguintes métodos são utilizados para acessar cada uma das facilidades do CommonDialog Control : ShowColor, ShowFont, ShowHelp, ShowOpen, ShowPrinter,  e ShowSave. 

Vamos exemplificar o uso desse controle com o programa abaixo. Como você verá, o “Botão de Opção” é um elemento de um array de seis botões, que serão carregados em tempo de execução. Portanto, após inserir esse botão no formulário, setar sua propriedade “index” para o valor 0 (zero).

Desenhe o seguinte formulário, com o respectivo código abaixo:  

Private Sub Form_Load()

   Dim i As Integer

   For i = 1 To 5

       Load Option1(i) ' Coloca cinco botões de opção na tela.

       Option1(i).Top = Option1(i - 1).Top + 350

       Option1(i).Visible = True

   Next i

   Option1(0).Caption = "Open" ' Põe caption em cada botão de opção.

   Option1(1).Caption = "Save"

   Option1(2).Caption = "Color"

   Option1(3).Caption = "Font"

   Option1(4).Caption = "Printer"

   Option1(5).Caption = "Help"

   Option1(0) = True

   Command1.Caption = "Click Aqui"   ' Coloca caption no botão de comando.

End Sub

Private Sub Command1_Click()

    If Option1(0).Value Then    ' Botão de opção "Open" foi selecionado

        CommonDialog1.ShowOpen

    ElseIf Option1(1).Value Then             'ou

        CommonDialog1.ShowSave  ' Botão de opção "Save" foi selecionado

    ElseIf Option1(2).Value Then             'ou

        CommonDialog1.ShowColor ' Botão de opção "Color" foi selecionado

    ElseIf Option1(3).Value Then             'ou

                                ' Botão de opção "Font" foi selecionado

        CommonDialog1.Flags = cdlCFBoth  'Setar esta constante antes de

                                         'executar o método ShowFont

        CommonDialog1.ShowFont

                                                                   'ou

    ElseIf Option1(4).Value Then 'Botão de opção "Printer" foi selecionado

        CommonDialog1.ShowPrinter             'ou

    ElseIf Option1(5).Value Then 'Botão de opção "Help" foi selecionado

        CommonDialog1.HelpFile = "VB.HLP"

        CommonDialog1.HelpCommand = cdlHelpContents

        CommonDialog1.ShowHelp

End If

End Sub

Exercício: Para verificar a utilização e algumas propriedades importantes do controle CommomDialog, implementar em seu programa os exemplos descritos a seguir:

Métodos ShowOpen e ShowSave:

Filter - Especifica o filtro para os arquivos de entrada, na forma: 

           descrição1|filtro1|descrição2|filtro2...

exemplo:CommonDialog1.Filter=“Todos os arquivos (*.*)|*.*|Arquivos de texto (*.txt)|*.txt|”

Filename - Contém o nome do arquivo a abrir ou gravar.

exemplo: Msgbox “Arquivo escolhido : ” + CommonDialog1.FileName

Método ShowColor: 

Color - Retorna a cor selecionada.

exemplo: Form1.Backcolor=CommonDialog1.Color

Método ShowFont: Verifique as propriedades, acompanhando o exemplo abaixo:

Private Sub Command1_Click ()

   CommonDialog1.Flags=&H3& ‘Disponibiliza os tipos de fontes de tela e impressora.

   CommonDialog1.ShowFont

                    ‘Atribui as propriedades do fonte, de acordo com a seleção escolhida

Text1.Font.Name = CommonDialog1.FontName

Text1.Font.Size    = CommonDialog1.FontSize

Text1.Font.Bold   = CommonDialog1.FontBold

Text1.ForeColor = CommonDialog1.Color

End Sub

Método ShowPrinter: Verifique as propriedades, acompanhando o exemplo abaixo

Private Sub Command1_Click ()

Dim Iniciopagina, Fimpagina, Copias, i as integer

CommonDialog1.ShowPrinter

                    ‘Configura a impressora, de acordo com a seleção escolhida

Iniciopagina = CommonDialog1.FromPage

Fimpagina = CommonDialog1.ToPage

Copias = CommonDialog1.Copies

For i=1  to Copias

   ‘ Coloque aqui o codigo para mandar os dados para sua impressora.

next i

End Sub.

Obs.: O método ShowHelp não invoca uma caixa de diálogo mas mostra na tela um arquivo de Help, cujo nome está especificado na propriedade “HelpFile”.  Arquivos de Help possuem a extensão “.HLP”, sendo que para se criar um arquivo de Help, há necessidade do uso de um compilador de Help.

Criação e uso de arquivos de Help, serão vistos a seguir.

CRIAÇÃO E USO DE HELP EM VB.
As aplicações desenvolvidas para Windows possuem geralmente um sistema de ajuda ao usuário (help), que pode ser acessado diretamente por meio de uma tecla ou por um item de menu, especialmente designado para esse fim. 

A maneira mais utilizada é a criação de um arquivo de help, que seja sensível ao contexto. O help desenvolvido dessa maneira, será apresentado quando o usuário pressionar a tecla F1. Nesse caso o help mostrará o tópico referente ao objeto que está com o foco no momento em que se pressionou F1.

Para criar o arquivo de help e incorporá-lo a uma aplicação em VB, seguem-se os seguintes passos:

-Criação do arquivo fonte do help (arquivo com extensão “.RTF” - Rich Text File);

-Criação do arquivo de projeto do help (extensão “.HPJ”);

-Compilação do arquivo fonte com base no arquivo de projeto, gerando o arquivo de help (extensão “.HLP”);

-Conexão do arquivo de help com a aplicação em VB.

Vamos detalhar cada um dos passos acima, por meio do exercício a seguir. Vamos incrementar o uso do controle “Common Dialog” utilizando a facilidade que esse controle possui de chamar e exibir arquivos de help. Em primeiro lugar, vamos criar o nosso arquivo fonte de help, por meio do MS Word.

Para isso, entre no Word e crie em seu disco o arquivo Ajudame.rtf, contendo o texto que tem início após a linha pontilhada a seguir. 

........................................................................................................................................

Título da Janela Principal do Arquivo de Help 

Esta será a primeira janela a ser apresentada quando o usuário acionar o sistema de help.


Texto1 de ajuda de janelaTEXTO1

Texto2 de ajuda de janelaTEXTO2

...................................Quebra de página.........................................................

Título da janela do Texto 1

Este texto ( Texto1)  será apresentado quando o usuário clicar na frase  “Texto1 de ajuda de janela”, na janela principal do sistema de help deste exercício, ou quando acionar a tecla F1 e esta janela estiver associada ao controle que estiver com o foco naquele momento.
................................................Quebra de página.........................................................

Título da janela do Texto 2
Este texto ( Texto2)  será apresentado quando o usuário clicar na frase  “Texto2 de ajuda de janela”, na janela principal do sistema de help deste exercício, ou quando acionar a tecla F1 e esta janela estiver associada ao controle que estiver com o foco naquele momento.
Esta janela será também usada para demonstrar a possibilidade de apresentação de um texto explicativo, que ficará sobreposto à janela de help. Isso ocorrerá quando o usuário clicar na palavra-chavePALAVRACHAVE deste texto.
..................................................Quebra de página.........................................................

Texto Explicativo.... Este texto será apresentado quando o usuário der um click na frase “palavra-chave”da janela do “ Texto 2”.

Este texto será sobreposto à janela em apresentação.

.....................Final do Texto.

................................................Quebra de página.........................................................

Obs.      É necessário para finalizar o documento, uma quebra de página e um espaço.

             A quebra de página pode ser obtida com a combinação de teclas “Ctrl + Enter”.

Formatação do Texto da Janela Principal:

As frases “Texto1 de ajuda de janela” e  “Texto2 de ajuda de janela” devem ser na cor verde e com duplo sublinhado, para podermos saltar à janela que contém o tópico desejado.

As palavras “TEXTO1” e “TEXTO2”, que vêm sem espaço logo a seguir, devem ser ocultas. Serão utilizadas para amarração com as janelas de Texto 1 e de Texto 2.

Formatação de Texto 2

A frase “palavra-chave”deve ser na cor verde e com sublinhado simples, para podermos chamar um texto que ficará sobreposto a esta janela.

A frase “PALAVRACHAVE” deve ser oculta para fazermos a amarração com o texto que queremos exibir.

Amarração das telas do arquivo de help: É feita por meio das palavras chave que estão em formato oculto, relacionando-as com a inclusão de notas de rodapé, nas telas em que quermos fazer a amarração.

A tela principal é a primeira tela a surgir quando chamamos o help e não é chamada por nenhuma tela. Mesmo assim, devemos incluir uma nota de rodapé. Faça isso, incluindo a nota de rodapé “AUXILIO” por exemplo.

Para isso, posicione o cursor no início do cabeçalho na página da tela principal, dê um click no item de menu “Inserir”, abra o item “Notas” e digite o símbolo “# “ na caixa de texto da janela apresentada.  Pressione o botão “Ok”.

Note que o Word abriu uma janela e já inseriu o símbolo “#” no início da linha. Digite agora a palavra “AUXILIO” nessa janela e em seguida feche essa janela.

Em seguida posicione o cursor no início do cabeçalho da janela do “Texto 1” e insira uma nova nota de rodapé com o símbolo “# “.

A seguir digite a palavra chave, sendo desta vez, a palavra “TEXTO1”, para fazer a amarração com a palavra “TEXTO1”  que está em formato oculto na janela principal.

Dessa maneira, já fizemos a amarração da janela principal com o Texto 1. Vamos fazer o mesmo para o Texto 2. Novamente, posicione o cursor no início do cabeçalho da página do Texto 2 e insira a nota de rodapé “#”. Após clicar em “Ok”, digite a palavra chave “TEXTO2” e feche essa janela.

Repita o procedimento para a página do Texto Explicativo, incluindo a nota de rodapé “#” e digitando a palavra “PALAVRACHAVE”. 

Obs. Existem outros símbolos para nota de rodapé, que possuem outras funções, tais como:

$: dá um título à página atual de help

k: fornece uma palavra chave para pesquisa. 

Quando salvar esse arquivo, não esqueça de salvá-lo usando o formato do word “.RTF”. Para isso, selecione  “Arquivo/Salvar Como”, abra a caixa “Salvar arquivo como:” e selecione “Formato Rich Text (RTF)”.
Criação do arquivo de projeto do help:

A seguir, vamos executar o programa que faz a compilação do texto que acabamos de criar, gerando um arquivo de help, que possui a extensão “.HLP”

O programa que faz a compilação, gerando o arquivo de help, tem o nome HCW.EXE e pode ser encontrado em um dos seguintes caminhos: 

C:\Arquivos de Programas\Borland\Delphi2.0\HELP\Tools

C:\Arquivos de Programas\DevStudio\VC\Bin

Localize o programa HCW.EXE e execute-o pela linha do DOS ou pelo “Executar...” do Windows.

A partir da tela apresentada, vamos criar o arquivo de projeto, que tem a extensão “.HPJ”. Esse arquivo irá conter as especificações que serão utilizadas para a compilação do arquivo “AJUDAME.RTF”, visando a geração do arquivo de help (Ajudame.hlp).  Nosso arquivo “Ajudame.HPJ” será composto pelas seções [OPTIONS], [FILES], [MAP] e [WINDOWS], devendo ficar parecido com a tela a seguir:

Para a criação das seções da tela anterior contendo essas informações,  proceda da seguinte maneira:

· Dê um click no item File/New, escolha o item “Help Project” e dê um click em OK;

· Na tela apresentada, escolha o drive e o nome do arquivo de projeto. Em nosso caso, é “a:ajudame”. A extensão .HPJ é colocada automaticamente. Note que ao pressionar o botão SALVAR, já é criada a tela com a seção [OPTIONS];

· Dê um click no botão “Options” e em “Help title:” digite o título “Meu Primeiro Help”. Esse título irá aparecer na barra de título da janela de help. 

· Para criar a seção [FILES], dê um click no botão “Files...” e em seguida dê um click no botão “Add”. A seguir, selecione o arquivo fonte para o compilador de help ( AJUDAME.RTF). Dê um click no botão “Abrir”,e em “OK”;

· A criação da seção [MAP] é feita pressionando-se o botão “Map...”. Para inserir os strings com os tópicos e o valor para amarração com o “HelpContextId” do controle da aplicação em VB, pressione o botão “Add...” para cada item e coloque os tópicos de acordo com a tabela abaixo: 

TOPIC ID
MAPPED NUMERIC VALUE
COMMENT

AUXÍLIO
10
PRIMEIRA TELA

PALAVRACHAVE
40
TEXTO PARA A PALAVRA CHAVE

TEXTO1
20
PRIMEIRO TEXTO

TEXTO2
30
SEGUNDO TEXTO

· Para finalizar, dê um click no botão “Window”, click no botão “Add”, escolha um nome para a janela do help aceite os valores “default” oferecidos e dê um click no botão “Ok”.

· Para gravar o arquivo de projeto de Help e fazer a compilação, dê um click no botão “Save and Compile”. Aguarde alguns instantes até terminar a compilação. Após a compilação, verifique se não há mensagens de erro. Se houver, corrija o problema e recompile o arquivo. Mesmo havendo mensagens de advertência (Warning), o arquivo “ajudame.hlp” deve ter sido gerado em seu disquete. 

· Verifique que em seu disquete devem ter sido gerados os arquivos “ajudame.hpj” e “ajudame.hlp”, sendo este último, o arquivo de help que usaremos em nossa aplicação.

Vamos agora fazer a conexão do arquivo de help com a aplicação em VB. Para isso, vamos utilizar a aplicação que possui o Common Dialog Control, incluindo o seguinte código no evento Form Load e incluindo o botão, tal como o exemplo a seguir:

Private Sub Form_Load()

App.HelpFile = "a:ajudame.hlp"

End Sub

Private Sub Cmdchamahelp_Click()

CommonDialog1.HelpFile = "a:ajudame.hlp"

CommonDialog1.HelpCommand = cdlHelpContents

CommonDialog1.ShowHelp

End Sub

Execute o programa e dê um click no botão “Chama help” para visualizar e navegar pelo Hipertexto do help.

Para implementaros o help sensivel a contexto nos  botões “Botão Help 20”, “Botão Help 30” e na caixa de texto, basta atribuir à propriedade HelpContextID dos objetos correspondentes, os valores que colocamos na seção MAP do arquivo de projeto do help (AJUDAME.HPJ).

Assim, se atribuirmos o valor 20 à propriedade  HelpContextID do botão “Botão Help 20”, quando pressionarmos a tecla F1 e esse botão estiver com o foco, será exibida a tela de help do Texto 1.

Exercício: Abra novamente o Word e utilize o símbolo “k” como nota de rodapé, fornecendo as seguintes palavras chave para pesquisa: 

-Primeira Janela

-Janela Texto 1

-Texto 2 - Janela

- Explicativo

Procedimento: Coloque o cursor no início do cabeçalho da janela principal e repita o procedimento de inserir notas de rodapé, porém utilizando agora  o símbolo  k  como nota de rodapé. Dê um click no botão “Ok”. Na janela que se abre, digite as palavras “Primeira Janela”. Repita esse processo para as demais páginas, usando as palavras da lista acima.

Recompile o arquivo, execute o programa em VB e verifique que agora está disponível para uso o botão “Localizar” da tela do Help.

USO DE APIs DO WINDOWS

O Windows fornece um grande conjunto de funções, tipos de dados e constantes que podem ser utilizadas por programas em VB. O conjunto desses recursos é chamado de API (Aplication Program Interface) e a possibilidade de sua utilização por programas em VB tornam a programação extremamente poderosa. As funções fornecidas pela API executam uma ampla gama de serviços tais como o acesso a recursos de periféricos, configuração de dispositivos do sistema, acesso a portas lógicas entre muitos outros. 

As “Functions”, “Types” e “Constant” fornecidas pela API não são nativas do Visual Basic, devendo portanto ser incorporadas ao programa que irá utilizá-las.  Para isso, existe o aplicativo “API Text Viewer” que facilita a utilização e inclusão de funções, tipos de dados e constantes no programa VB.

Para utilizá-lo vamos executar o “API Text Viewer” que nos mostra uma janela com duas divisões:

Divisão Superior: Contém a lista disponível de funções ou declarações de tipo de dados ou declaração de constantes. Devemos selecionar no item “API Types” qual o tipo que queremos.

Para fazermos um exercício, selecione “Declares” e abra o item “File” do menu.  Abra o item “Load Text File”e selecione o arquivo “Win32api.txt”. Note que após algum tempo de processamento, o espaço abaixo é preenchido com muitos protótipos de funções API.

Ao selecionarmos uma função e clicarmos em “ADD”, essa função (seu protótipo) é copiado para a divisão inferior. Faça isso para as funções ShowCursor e GetTickCount.

Divisão Inferior: Após selecionarmos  os itens que nos interessam e os enviarmos à divisão inferior da janela, clicamos o botão “Copy” para enviarmos os itens dessa janela à Área de Transferência.     

Após isso, podemos colar o protótipo das funções API em um “Module” de nosso projeto ou no “General” de algum formulário.

Em seguida, podemos usar a  função API, como uma função qualquer do Visual Basic. 

Exemplo: Vamos usar como exemplo a função “ShowCursor”, que tem por finalidade mostrar ou esconder o apontador do mouse.

Se o valor do argumento “bShow” for False(0), a função esconde o apontador do mouse. Se for True(-1), exibe o apontador. Na verdade, ela incrementa ou decrementa um contador interno. O apontador do mouse fica visível, se o contador interno de “ShowCursor” for  >= 0.

Para testar o uso de APIs, faça o programa abaixo. Inclua o protótipo da função ShowCursor, de acordo como explicado acima.

Os protótipos das funções API abaixo, devem ser declarados no “General”, precedidos da palavra “Private”: 

Private Declare Function ShowCursor Lib "user32" (ByVal bShow As Long) As Long

Private Declare Function GetTickCount Lib "kernel32" () As Long

Private Sub cmdvisivel_Click()

   Dim bShow As Long

   bShow = -1         'Faz bShow True

   ShowCursor (bShow) 'executa API Showcursor()

End Sub

Private Sub cmdInvisivel_Click()

   Dim bShow As Long

   bShow = 0    'Faz bShow false

   ShowCursor (bShow)  'Chama API do Windows

   cmdvisivel.SetFocus   'Põe foco no botão Visível para acionar com "Enter"

End Sub

Note que o cursor fica invisível apenas quando estiver sobre a aplicação que provocou sua invisibilidade. Quando deslocado para outra tela, ele torna-se novamente visível.

Para utilização da função API GetTickCount, faça o seguinte programa:

Private Sub Form_Click()

      Print GetTickCount

End Sub

A função API GetTickCount retorna um inteiro longo contendo o tempo em milisegundos, contados a partir do momento em que foi ativado o Windows.

Essa função é bastante útil para verificarmos o tempo consumido por um processo. 

Acompanhe o programa a seguir, que faz a contagem do tempo em milisegundos, gasto na realização do loop “For ... Next”.

Private Sub Form_Click()

  Dim inicio As Long, contador As Long

  inicio = GetTickCount

  For contador = 1 To 100000

  Next contador

  Print "Tempo decorrido = " + Str(GetTickCount - inicio)

End Sub

Pergunta: Se a execução deste trecho de programa é sempre igual, por que o tempo consumido em duas execuções geralmente é diferente?

Obs. Se seu equipamento possuir uma placa de som e você quiser executar um arquivo de som (por exemplo um arquivo com extensão “.WAV”) ao ocorrer um evento em seu programa em VB, use a função API “sndPlausound”, mais ou menos como abaixo:

Private Declare Function sndPlaySound Lib "winmm.dll" Alias "sndPlaySoundA" (ByVal 

                                                 lpszSoundName As String, ByVal uFlags As Long) As Long

Private Sub Form_Click()

Dim x As Long

sndPlaySound "c:\windows\msremind.wav", x

End Sub

Lembra do Relógio Despertador? Seria interessante se ao despertar, o programa ficasse em primeiro plano. Existe uma API que faz isso:  SetForegroundWindow (Form1.HWND).

Quer que seu programa dê uma pausa por um intervalo de tempo? Use a API   Sleep.

Quer colocar o ícone de seu programa na “Bandeja” (System Tray)? Use a API Shell_NotifyIcon .

Veja como fazer isso, consultando a página  http://www.silicio.com.br/vb/dicas/win_019.html

EXERCÍCIO:
A  empresa aérea  “VB AIR”, recentemente operando no setor de turismo, está promovendo o evento  “Bermudas View”, que é sua primeira excursão panorâmica sobre o famoso “Triângulo das Bermudas”.

Para efetuar o cadastramento e reservas de lugares para os intrépidos passageiros em seu avião, solicita que você desenvolva um sistema em Visual Basic, que faça no mínimo, o seguinte:

Mostre uma tela com os lugares já ocupados e os lugares disponíveis, para que a vítima escolha a poltrona que mais lhe agrade. Poderia ser algo como a seguinte tela:

Após o passageiro escolher e selecionar sua poltrona preferida, o operador dá um click no botão “Reserva Lugar”, para poder entrar com os dados cadastrais do interessado. Esses dados deverão ser gravados em um banco de dados, para que posteriormente seja emitido um relatório. 

A tela para cadastramento, poderá ser algo como a tela abaixo.

Obs.: Os dois botões apontados com (***) abaixo, são botões   “3D Command Button Control” , do grupo de controles “Sheridan 3D Control”.

Inclua o conjunto de controles “Sheridan...”, na sua ToolBox e selecione o ícone  [image: image59.png]


 para inseri-lo em sua aplicação.

Você pode dar uma aparência tridimensional ao botão e pode também inserir ícones ou desenhos feitos pelo PaintBrush na propriedade Picture desses controles.

Utilize o Help On Line (F1) para obter mais informações sobre esses controles.

  (***)

Caso seja pressionado o botão “Relatório”, deverá ser impresso um relatório em ordem crescente do número da poltrona.

Obs.: Você pode acessar campos de um banco de dados que esteja associado ao “Data Control”, da seguinte maneira: Data1.Recordset.Fields(“Poltrona”).Value=“Poltrona número 13”.

Obs.:

Temos  também outras maneiras de acessar os registros do banco de dados. Acompanhe o seguinte exemplo, onde acessamos o campo “POLTRONA”, utilizando outros médodos do “Data Control” : 

Private Sub Command2_Click()

Cadastra.Data1.Refresh  'Cria um novo "recordset"

Do Until Cadastra.Data1.Recordset.EOF 'executa até ser fim de arquivo

    Print Cadastra.Data1.Recordset("poltrona") 'imprime o campo "poltrona" do registro corrente

   Cadastra.Data1.Recordset.MoveNext   'move o ponteiro de arquivo p/ próx. registro

Loop

End Sub 

Para tornar seu programa “mais profissional”, inclua “PISTAS” nos botões da aplicação, utilizando a propriedade ToolTipText. 

Observações:


- cada formulário possui uma área denominada “general” onde se declaram variáveis que são estáticas e são globais a todas as procedures sub e procedures function SOMENTE DAQUELE formulário.





- cada módulo possui uma área “general” onde se pode declarar variáveis GLOBAIS AO PROJETO e novos tipos de variáveis.





- as variáveis declaradas dentro das SUBs e FUNCTIONs de cada formulário, são locais a esses formulários.





- variáveis declaradas na área “general”e as variáveis declaradas como “estatic” são inicializadas com o valor 0 e mantém os valores atribuídos a elas enquanto durar o processamento do programa.





-  variáveis declaradas com o tipo “variant” adaptam-se ao tipo do dado a elas atribuído, porém seu manuseio gasta mais tempo de processamento.

